
概要

これはMapleプログラミングのチュートリアルです.このチュートリアルの目的は,数値計算や線形代数を

行う簡単な Mapleプログラムを書いたり,記号式や数式を簡略化したり変換したりするプログラムを作る

方法を示すことです.電卓として対話しながらMapleを使うことに読者が慣れていることを前提にしており

ます.

1 はじめに

他のプログラミング言語に慣れている人への一言

� Mapleは手続き型プログラミング言語です.また,たくさんの関数型プログラミング機能も持っており

ます. BASIC, Pascal, Algol, C, LispあるいはFortranでプログラムを書いた経験があれば, Mapleの計

算プログラムを書くのはすぐに出来るでしょう.

� Mapleは Cや Pascalのように厳しく型を使っておりません. 型宣言は不用です. この点に関しては

Mapleは Basicや Lispによく似ております. しかし Mapleにも型は存在します. 型のチェックは実行

時になされるので,それを行うときははっきりとプログラムに書いておかなければなりません.

� Mapleは対話型で,そのプログラミング言語はインタプリタで解釈されます. このインタプリタの負

荷のために大規模な計算プログラムを実行するのに適しておりません.けれども高精度の数値計算を

行ったり,計算式を作る道具としては適切なのもです.

この文書はMapleバージョン V Release 2に基づいております.*1) Mapleの開発は現在も続いております.

新しいバージョンが1, 2年毎に出てきて, Mapleの数学的機能の変更だけでなく,プログラミング言語やユー

ザインターフェースの変更も行っております.こうした理由から,私はこの言語の現状の機能から将来なく

なる機能なども説明し,さらにMapleの将来のバージョンに取り込まれると思われることなども説明するこ

とにしました.

Maple V Language Reference Manualは Mapleプログラミングの主要な文献です. これは Springer-Verlag

から出版されております. ISBN番号は 0-387-87621-3です.他の役立つ情報源は, Waterloo Maple Softwears

から入手できます. Maple V Release 2の Release Notesや Springer-Verlagから出版されている First Leaves :

Tutorial Introduction to Maple,それに Mapleの中にある ? コマンドで呼び出せる膨大なオンライン文書で

す.この文書を書く理由の一つは,私がここで述べた重要ないくつかの事柄は,他に発表していなかったり文

書に埋もれてしまっていることが挙げられます.

1.1 評価

始めに, Mapleと従来のプログラミング言語とのもっとも重要な差異を指摘しておきます. 識別子に値が

割当てられていなければ,それはそれ自身を表します.これは記号 (symbol)と呼ばれてます.記号は方程式

の未知数,多項式の変数,積和の添数などを表すのに用いられます. 次の Maple割当て文を考えてみましょ

う.

> p := x ˆ 2 + 4 * x + 4 ;

p := x2+4x+4

ここでは識別子 p には数式 x2+4x+4が割当てられております.識別子 xには値が割当てられておりま

せん. これはまさに記号で未知数なのです. 識別子 pには値が割当てられております. こちらのほうはプロ

グラミング変数のようなもので,その値はちょうど通常のプログラミング変数のように次の計算で使われま

す. pの値は何でしょうか ?

> p ;

x2+4x+4

*1)(訳注)この文書のコマンドや出力は,すべて Maple V Release 5に改めてあります.プログラムの一部にエラーなどの修正を行なっ
たものもあります.

それは数式 x2+4x+4です.それでは xの値は何でしょうか ?

> x ;

x

それは記号 xです. 変数には記号を含む値を割当てることができるので,評価 (evaluation)の問題が起こっ

てきます.次の文を考えてみましょう.

> x := 3 ;

x := 3

> p ;

25

ここでは xに値 3を割当てて, Mapleに pの値を出力するように要請しました. Mapleはどんな値を出力

するか? 多項式 x2+4x+4を出力するのでしょうか,あるいは多項式を評価して 32+4�3+4を計算しそ

の答 25を出すのでしょうか? 見られるとおりMapleは後者をとりました.この評価の問題は時々刻々出て

きて,それがプログラムの効率や意味に影響を与えます. Mapleと従来のプログラミング言語との差異,すな

わち識別子はプログラミング変数と数学的変数の両方に用いられることはすばらしいことです.しかしこの

２つの混同に注意しなければなりません.ユーザが出逢ういろいろな問題点は識別子が記号とプログラミン

グ変数の両方に使えるということに関係があります. それでは,
R

pdxを計算しようとすればどんなことが

起こるでしょうか ?

> int ( p, x ) ;
Error, (in int) wrong number (or type) of arguments

積分関数 int にエラーが起こりました.ここでは積分変数 xは記号とみなされております.しかし前のほ

うで xには整数 3が代入されました. Mapleはこの引数を使って int 関数を評価するので, 25を 3に関し

て積分しようとしました. 3に関しての積分することはまったくナンセンスです! xを記号に戻すにはどう

するのでしょうか ? Mapleでは次のようにして変数 xを未割当て (unassign)とします.

> x := ’x’ ;

x := x

> int ( p, x ) ;

1
3

x3+2x2+4x

1.2 式 : 和,積,ベキ,関数

Mapleでは数式たとえば sin(x+π=2) や x3y3�2=3のようなものは,式 (expressions) と呼ばれておりま

す. これらは記号, 数字, 算術演算子および関数から成り立っております. 記号は sin, x, y, Pi のよ

うなものです. 数字は 12; 2
3;2:1などです. 算術演算子は + (加法), � (減法), � (乗法), = (除法)および ˆ (ベ

キ乗)です. 関数の例として sin (x), f(x,y), min (x1,x2,x3,x4) があります. たとえば,多項式

p= x2y+3x3z+2はMapleでは次のように入力されます.

> p := xˆ 2 * y + 3 * xˆ 3 * z + 2 ;

p := x2 y+3x3z+2

そして式 sin(x+π=2)e�x は次のように入力されます.

> sin( x+Pi/2 ) * exp( -x ) ;

cos(x) exp(�x)

Mapleは sin(x+π=2)を cos(x)と簡単化したことに注意して下さい. Mapleの式は数式木 (expression trees)す

なわちコンピュータ用語でいう DAG (非巡回的有向グラフ )で表されます.我々が式を使うためにMaple関

数のプログラムを書くとき,基本的には計算木を扱っていることになります.この計算木を調べる 3つの基

本的ルーチンが type, op および nops です. type 関数

type( f ; t )

は式 f が型 t ならば値 trueを返すものです. 基本の型は string, integer, fraction, float,

‘+‘, ‘ �‘, ‘ˆ ‘, function です. whattype 関数も式の型を出力するのに役立ちます.たとえば,我々

の多項式 pは 3項の和です.したがって

> type ( p, integer ) ;

f alse

> whattype( p ) ;

+

> type ( p, ‘+‘ ) ;

true

このバッククォート文字 ‘ の使い方に注意して下さい. Mapleではバッククォートは, / , . などの奇妙な文

字を含む文字列に使用されます.バッククォートはフォワードクォート (アポストロフィ ) ’ やダブルクォー

ト " と同じものではありません.

他の 2つの数値型が rational と numeric です.型 rational は有理数すなわち整数と分数を指しま

す. 型 float は浮動小数点数すなわち小数点を含む数字を指します. 型 numeric はこれらの数字のいず

れか,すなわち型 rational か float の数字を指します. MapleユーザはMapleが有理数と近似値を区別

していることに注意して下さい.小数点が存在することが重要なのです! 次の文を考えて下さい.

> 2/3 ;

2
3

# That is a rational number, the following is a floating point number

(これは有理数で,次の数字は浮動小数点数です )

> 2 / 3.0 ;

:6666666667

技術者に注意 : あなたがたの大部分の人は, Fortranや Cのソフトウェアを使ったりプログラ

ミングしているでしょう. Fortranや Cでは常に小数点を用います. すなわち 1=2の代わり

0:5と書くように学んでいるでしょう. Mapleでは,式をタイプするときはたぶん 1=2とする

でしょう.データをタイプするときはたぶん 0:5を用いるでしょう. Mapleでは 1=2と 0:5は

同じものではありません.

我々の例の pはいくつかの項の和です. これはいくつの項を持っているでしょうか? nops 関数は式のオ

ペランドの個数 (number ofoperands)を返します.和の場合はそれに含まれる項数になります.積の場合は

それに含まれる項数です.したがって,次のようになります.

> nops( p ) ;

3

op 関数は式中のオペランド ( operand)を 1つ抽出するために用いられます. i が 1から f のnops までの

範囲にあるとき,次の構文

op( i; f )

は式 f の i 番目のオペランドを抽出するものです. 我々の例では,これは和 f の i 番目の項を抽出すること

になります.

> op( 1, p ) ;

x2 y

> op( 2, p ) ;

3 x3 z

> op( 3, p ) ;

2

op 関数は次のようにも用いられます.

op( i:: j; f )

これは f の i から j までのオペランドの列を返します.たとえば,次のようになります.

> op( 1..3, p ) ;

x2 y; 3x3 z; 2

役立つ短縮形は op( f ) で,これは op( 1..nops( f ), f ) と同じで, f のすべてのオペランドの列を作り

ます.和 pの第 2項は何でしょうか ? これは 3つの因子の積です.

> type ( op ( 2, p ), ‘ * ‘ ) ; # It is a product (これは積です )

true

> nops ( op ( 2, p ) ) ; # It has 3 factors (これは 3因子を持っています )

3

> op ( 1, op ( 2, p ) ) ; # Here is the first factor (これは第一因子です )

3

> op ( op ( 2, p ) ) ; # Here is a sequence of all 3 factors

(これは 3因子のすべての列です )

3; x3; z

添数付き記号名と関数

Mapleは 2種類の変数を持っていて,それを記号名 (names)と呼んでおります.それは型 string を持つ

x, sin, Pi のような文字列と,型 indexed を持つ A1; Ai; j ; Ai j のような添数付き記号名すなわち添字付

き変数です. これらの例は Mapleでは A[1], A[i,j], A[i][j] と入力されます. Mapleのほとんどの関

数はこの 2種類の変数を受け入れます. Mapleの型である nameは文字列か添字のいずれかを指します.

例

> type ( a, string ) ;

true

> type( a, name ) ;

true

> whattype ( A[1] ) ;

indexed

> type( A[1], indexed ) ;

true

> type ( A[1], name ) ;

true

f が添数付き記号名のとき, nops 関数は添数の個数を返し, op 関数は i 番目の添数を返します.

また op( 0, f )は添数の記号名を返します.

例

> nops( A[i,j] ) ;

2

> op ( 1, A[i,j] ) ;

i

> op ( 0, A[i,j] ) ;

A

> nops( A[i][j] ) ;

1

> op( 1, A[i][j] ) ;

j

> op ( 0, A[i][j] ) ;

Ai

関数は添数付き記号名とよく似た動作をします.関数の呼び出しの構文は次のとおりです.

f (x1;x2; � � � ; )

ここで f は関数の記号名で, x1; x2; � � � は引数です. nops 関数は引数の個数を返し, op 関数は i 番目の引数

を返します.また op ( 0, f )は関数の記号名を返します.

例

> nops ( f( x, y, z ) ) ;

3

> op ( 1..3, f ( x, y, z ) ) ;

x; y; z

> op ( 0, f ( x, y, z ) ) ;

f

これでMapleの式を作ったり,一部を抽出したりすることができるようになります.練習問題6は Mapleの

式の一部を抽出するMapleプログラムを書く問題です.ここではいくつかの例を使って,それを対話的にど

のように行うかを示して終わりとしましょう.

> f := sin( x[1] ) ˆ 2 * ( 1-cos( Pi * x[2] ) ) ;

f := sin(x1 )
2 (1 � cos(π x2 ))

> whattype ( f ) ;

�

> op ( 1, f ) ;

sin(x1)
2

> op ( 1, op ( 1, f ) ) ;

sin(x1 )

> op ( 1, op( 1, op ( 1, f ) ) ) ;

x1

1.3 文 : 割当て文,条件文,繰返し文

Mapleの割当て文, if 文, for文, while文の構文は Algol 60から採ったものです. 割当て文は次のように

なっています.

name:= expr

ここで exprは任意の式で, nameは変数記号名です.ここで,いずれ出てくる評価問題について述べておきま

しょう.この問題が生じるのは,変数に数字を割当てることができるように,記号を含む式にも変数を割り当

てることができるからです.前にやった pへの割当て文を考えてみましょう.

> p := x ˆ 2 + 4 * x + 4 ;

p := x2+4x+4

この場合は記号名 pに数式を割当て,その数式は記号 xを明示的に含んでおります.それでは同じ記号名を

割当て文の両辺で使ったらどうなるでしょうか?

> p := p + x ;

p := x2+5x+4

何ら変わったことは起こりません. それではここの評価規則はどのように適用されたかを見てみましょう.

記号名 pには右辺 p+ xを評価し,さらに簡単化した結果が割当てられています. p+ xを評価するとき, p

の値は多項式 x2+4x+4で xの値はまさに xそのものです. したがって評価結果は式 (x2+4x+4) +xで

す. 次にこれが簡単化されて最終結果 x2+5x+4になりました. それでは難題を出しましょう. pに事前に

値が割当てていなかったらどうなるでしょうか? 見てみましょう. pの代わりに qを用いると次のように

なります.

> q := q + x ;
Warning, Recursive definition of name

q := q+x

今度は明らかにユーザが最初に, qにある値を割当てたことを忘れていたとしましょう. これはあり得る

ケースです.それでは続けてMapleにどんなことが起こるか見てみましょう.前の例のように qが xのよう

な値を持つことを Mapleは絶対許しません.

上の文はMapleの警告をもたらしました.記号名 qには今度は qを含む数式が割当てられました. qを評

価しようとすれば今度はどうなるでしょうか ? q+xを評価しなければなりません. しかし q+xを評価す

るためには,また qを評価しなければなりません. ここには無限の評価の繰返しが起こります. 筆者の研究

室のシステムでは, qを評価しようとすると,特別なクラッシュが起こりMapleは死んでしまいました.

> q := q + x ˆ 2 ;
Error, too many levels of recursion

このような警告から回復するには,単に変数 q を未割当てにする,すなわち, q := ’q ’ ; とします. 一

般に大変コストがかかるので, Mapleはすべての再帰的割当てを検出しないことに注意して下さい.

注意 : 上の例の q のような記号名の再帰的定義は,評価の際 Mapleがスタック領域をはみ出

したときクラッシュするのが普通です.システムによってはMapleは死んでしまいます.他の

システムでは,死ぬ前に問題があると報告して停止します.読者の所で Mapleがクラッシュし

たことがあれば,いま述べたことが理由である見込みが大いにありえます.アルゴリズムの中

に本当の無限ループがあれば,もちろん Mapleは死んでしまいます.

通常のプログラミング言語ではすべての変数は値を持っていますから,この再帰的定義の問題は起こりませ

ん.値が割当てられていなければ,間違いなくエラーです.言語にもよりますが,変数はデフォルト値をとる

か,その時点でたまたまメモリにあった適当な値をとるか,あるいはまれな場合ですが言語がそれを検出し

てエラーを出そうとします.

Mapleの条件文は次の構文をとります.

if expr then statseq

[ elif expr then statseq] �

[ elif statseq]

fi

ここで statseqはセミコロンで区切られた文の列で, [ ] はオプションの部分を表し, � はゼロ回以上反復され

る部分を表します.典型的な if 文は次のとおりです.

if x < 0 then -1 elif x = 0 then 0 else 1 fi

for文は 2つの構文を持っております.第 1構文は

[ for name] [ from expr] [ by expr] [ to expr] [ while expr]

do statseqod

したがって for, from, by, to, while の各節は省略可能です. 省略された場合は,そのデフォルト値はそれぞれ

ダミー変数, 1, 1,∞ , trueとなります. for ループの典型的な例は次のとおりです.

for i to 10 do print( iˆ 2 ) od ;

for, from, by および toの節を省略すれば,いわゆる whileループになります.

i := 10ˆ 10+1 ;

while not isprime ( i ) do i := i + 2 od ;

print ( i ) ;

for ループと whileループを組み合わせるとすばらしいものになることがあります. たとえば, 1010より大

きい最初の素数を探すこの例は次のように書けます.

for i from 10ˆ 10+1 by 2 while not isprime ( i ) do od ;

print ( i ) ;

上の例は繰返しが終了したあとでも繰返しの指標の値を呼び出すことができることを示しております.

for 文の第 2構文はいわゆる for-in 文です.これは頻繁に起こる繰返しにまさにうってつけのものです.

for i to nops ( s ) do f ( op ( i,s ) ) od ;

ここで sは和でもよいが,一般にはMapleの式あるいはリストや集合のようなデータ構造 (これについては

次節を参照して下さい)でもよいのです.これは for-in 文を使うと次のように書けます.

for i in s do f ( i ) od ;

for-in 文の構文は次のとおりです.

[ for name] [ in expr] [ while expr]

do statseqod

2 データ構造

非常に複雑なプログラムでは,データの操作や保存を行ったりします.データの表現法は,我々の書くアル

ゴリズムやプログラムの実行速度に影響します. Mapleには良いデータ構造がいくつかあります.ここで調

べるのは,式列,リスト (あるいはベクトル),集合,表 (あるいはハッシュ表),配列です. Mapleにはレコード

や結合型リストはありません.これらがMapleでどのように実現されるかについては,この節の終わりで述

べます.

2.1 式列

式列とはコンマで区切られた式の列のことです.たとえば,

> s := 1, 4, 9, 16, 25 ;

s := 1; 4; 9; 16; 25

> t := sin, cos, tan ;

t := sin; cos; tan

となります.式列の式列は 1つの式列にまとめられます.すなわち,式列は結合的 (associative)です.たとえ

ば,

> s := 1, ( 4, 9, 16 ), 25 ;

s := 1; 4; 9; 16; 25

> s, s ;

s := 1; 4; 9; 16; 25; 1; 4; 9; 16; 25

となります.特別な記号である NULL は空の式列として用いられます.式列はいろいろな目的に用いられま

す.次の節では式列からリストや集合がどのように作られるかを示します.ここでは関数呼び出しは実際に

式列から作られることに注意しておきます. たとえば, Mapleの min 関数と max関数は引数として任意個

数をとること,すなわち引数の式列を受付けます.

> max ( s ) ;

25

> min ( s, 0, s ) ;

0

注意： op 関数と nops 関数は式列に適用できません.その理由は式列自身が関数の引数になる

からです.したがって,呼び出し op ( s ) ; と nops( s ) ; はそれぞれ op(1, 4, 9, 16, 25)

と nops(1, 4, 9, 16, 25) と同値で,これはエラーとなります. op や nops を使用したけ

れば,あらかじめ式列をリストに直して下さい.

seq 関数は式列を作るのに非常に役に立ちます. 2種類の for 文に対応して, 2種類の使い方があります.

最初の 1つは次のとおりです.

seq( f (i); i = m: :n)

seq の動作はちょうど次の繰返し文をプログラムしたようになります.

s := NULL ;

for i from m by 1 to n do s:=s, f(i) od ;

たとえば次のようになります.

> seq ( iˆ 2, i=1 . . 5 ) ;

1; 4; 9; 16; 25

> s := NULL ; for i from 1 to 5 do s:=s, iˆ 2 od ;

s :=

s :=1

s :=1; 4

s := 1; 4; 9

s := 1;4; 9; 16

s := 1; 4;9; 16; 25

seq は中間の式列を作らないので for文より効率が良いことに注意して下さい. seq の他の使い方は

seq( f (i); i = a )

で,これは次のものと同値です.

seq ( f ( op( i, a ) ), i = 1 . . nops ( a ) )

ここに 2つの面白い例があります. coeff 関数は xに関する多項式の i 次の項の係数を求めるものです.

Mapleの D関数は微分演算子です.

> a := 3 * xˆ 3 + y * x - 11 ;

a := 3x3+yx�11

> seq ( coeff ( a, x, i ), i=0 . . degree( a, x ) ) ;

�11; y; 0; 3

> seq ( D ( f ), f = [ sin, cos, tan, exp, ln ] ) ;

cos;�sin; 1+ tan2; exp; a ! 1
a

2.2 リストと集合

リスト,集合,関数は式列から作られます. リストはオブジェクトをまとめたデータ構造です. 角括弧はリ

ストを作るときに用いられます.たとえば次のようになります.

> l := [ x, 1, 1-z, x ] ;

l := [x; 1; 1�z; x]

> whattype ( l ) ;

list

空のリストは [ ]で表します. 集合もオブジェクトをまとめたものです. リストと集合の差異は,集合から重

複したものが除かれるということです.大括弧は集合に用いられます.たとえば次のとおりです.

> s := f x, 1, 1-z, x g ;

s := f1; x; 1�zg

> whattype ( s ) ;

set

空集合は f gで表します. nops 関数はリストや集合の要素の数を返し, op 関数は i 番目の要素を抽出しま

す.式列,リスト,集合の i 番目の要素を呼び出すのに,添数記法を使うこともできます.たとえば,次のとお

りです.

> op ( 1,s ) ;

1



14 Mapleプログラミングの基礎

> s [ 1 ] ;

1

> op ( 1 . . 3, s ) ;

1; x; 1�z

> s [ 1 . . 3 ] ;

f1; x; 1�zg

次の繰返し文は,リストや集合が要素 xを含むときに trueを出力し,そうでないときに falseを出力します.

for i to nops(s) while s[i]<>x do od ;

if i > nops(s) then print(false) else print(true) fi ;

member関数を使ってもこれと同じことができます. member( x, s) は要素 xがリストあるいは集合 sに

属していれば, trueを返します.次のようにすると,リスト l に新しい要素を追加できます.

l := [ op ( l ), x ] ;

subsop 関数を使いリストから i 番目の要素を削除できます.

l := subsop( i=NULL, l ) ;

この subsop 関数は任意の型の式に適用できます.集合演算子は union, intersect, minus で,たとえ

ば次のとおりです.

> t := f x, z, u g ;

t := fx; z; ug

> s union t ;

fx; z; 1�z; ug

Mapleユーザが多分注目することは, Mapleが集合の要素を予測できない奇妙な順序で並べることでしょ

う. 重複した要素を削除したり,集合の和や積や差を作ったりするするときに Mapleが用いるアルゴリズ

ムは,入力した集合の要素を最初に整列することからすべて始まります. Mapleはマシンアドレス (machine

address)に従って,すなわち要素がコンピュータ内のメモリにある順序で整列します.このことは要素がメ

モリ内の置かれている場所に左右されるので,その順序は奇妙で予測できないものになるのです. Mapleの

集合はこのように実装されているので,集合演算を非常に速く行うことができます.

2.3 表

表やハッシュ表は効率的表を書くのに非常に役立ちます.表とは離散的なデータ間の 1対多関数です.た

とえば,ここに英語の色名のフランス語とドイツ語の訳語の表があります.

> COLOUR [ red ] := rouge, rot ;

COLOURred := rouge; rot

> COLOUR [ blue ] := bleu, blau ;

COLOURblue := bleu; blau

> COLOUR [ yellow ] := jaune, gelb ;

COLOURyellow := jaune; gelb

COLOUR表の定義域は英語で書かれた色名です. この表の値域はフランス語とドイツ語で書かれた色名の

式列です.一般に,定義域と値域の値は 0個あるいはそれ以上の値の式列です.表の定義域の値はキー (key)

とか指標 (indices)と呼ばれております. Mapleの indices 関数はこの式列を返します.表の値域の値は値

(value)とかエントリ (entries)と呼ばれております. Mapleの entries 関数はその式列を返します. たと

えば次のとおりです.

> indices ( COLOUR ) ;

[ red]; [yellow]; [blue]

> entries ( COLOUR ) ;

[ rouge; rot ]; [ jaune; gelb]; [bleu; blau]

indices 関数と entries 関数が表の指標とエントリを返す順序は,必ずしもそれが表に入力された順序

と同じではありません.これはMapleがハッシングを用いているからで,その結果探索は非常に速くなりま

すが,エントリが作られた順序は失われました.しかし, indices 関数と entries 関数の出力は 1対 1に

対応しております.

表を操作するにはどうするでしょうか? 表のキーすなわち指標が与えられると,対応するエントリはす

ばやく探すことができます.すなわち次の操作は

> COLOUR [ red ] ;

rouge; rot

色 redのフランス語名とドイツ語名を返します.どの位速いでしょうか? 表の大きさにかかわらず,処理時

間は一定です. 1000の異なる色を表が持っていても,その表を探索する時間は非常に速いのです. これが表

の基本的な使用法です.情報は割当て文で表に入力され,表探索は表の添数を用いて行われます.これ以外に

表の操作があるでしょうか? assigned 関数を用いてあるエントリが表にあるかどうかを検査できます.

そして,未割当て文によって表からエントリを削除できます.たとえば次のようになります.

> assigned ( COLOUR [ blue ] ) ;

true

> COLOUR [ blue ] := ’COLOUR [ blue ]’ ;

COLOURblue := COLOURblue

> assigned ( COLOUR [ blue ] ) ;

f alse

> print ( COLOUR ) ;

table( [

red = ( rouge; rot )

yellow= ( jaune; gelb)

] )

2.4 配列

1次元配列は次の array コマンドを用いて作られます.

array( m::n ) ;

これは指標 m; m+1; � � � ; nをもつ配列を作ります.エントリは,表のときと同じように割当て文で配列に挿

入されます.たとえば次のようになります.

v := array ( 1..n ) ;

v [ 1 ] := a ;

for i from 2 to n do v[ i ] := a*v[ i-1 ] mod n od ;

1次元配列は指標を 1個持っている表と似ていますが,こちらの指標は固定された範囲内の整数に限定さ

れております. 配列は表よりも効率よい呼び出しができて,指標の範囲検査もできます. ここに数列を整列

するために使用された 1次元配列の例があります. たとえば,上と同じ配列 v =array ( 1..n ) が与えら

れているとします.ここに書かれたコードは,バブルソートを用いて vを昇順に整列するものです.

for i to n - 1 do

for j from i + 1 to n do

if v[i] > v[j] then temp := v[i] ; v[i] := v[j] ; v[j] := temp fi

od

od ;

1次元配列の別の応用は中間的なデータ構造に使うのもです.たとえば,多項式 a(x) = ∑m
i=0 ai xi を係数

のリスト [a0; � � � ; am]で表すとします.次数 nの多項式 bも与えられているとします. 配列 cを用いて次の

ように積 a�bを求めます.

m := nops(a) - 1 ; # degree of a

n := nops(b) - 1 ; # degree of b

c := array(0..m+n) ; # allocate storage for the product

for i from 0 to m + n do c[i] := 0 od ;

for i from 0 to m do

for j from 0 to n do

c[i+j] := c[i+j] + a[i+1] � b[j+1]

od

od :

[ seq(c[i] ,i=0..n+m)] ; # put the product in a list

2次元配列およびもっと高次の配列も同じように動作します. 2次元配列は次のコマンドで作られます.

array(c::d; m::n) ;

1例として, 3変数 x1; x2; x3の対称多項式から成るベクトルが与えられているとします.

v := array ( 1..4 ) :

v [ 1 ] := 1 :

v [ 2 ] := x [ 1 ] + x [ 2 ] + x [ 3 ] :

v [ 3 ] := x [ 1 ] � x [ 2 ] + x [ 1 ] � x [ 3 ] + x [ 2 ] � x [ 3 ] :

v [ 4 ] := x [ 1 ] � x [ 2 ] � x [ 3 ] :

Mi; j を vi の xj に関する導関数とする 2次元配列M を作ってみましょう.

> M := array ( 1..4, 1..3 ) :

> for i to 4 do for j to 3 do M[ i,j ] := diff ( v[ i ], x[ j ] ) od od :

> M ;

M

> eval ( M ) ;
2
66664

0 0 0

1 1 1

x2+x3 x1+x3 x1+x2

x2x3 x1 x3 x1 x2

3
77775

注意 : Mの値は配列 Mの名前そのものであることに注意して下さい. 配列と表の評価規則は

特殊なものです. その理由は未割当てエントリーをもつ配列が使えるようにしているためで

す.ここではこれについて詳細に述べません.当分の間,配列や表を出力したり,手続きから配

列や表を返すときはいつも eval 関数を使って下さい.

arrayに関する詳細な情報が欲しいときは, ? array で調べて下さい. また添数 1から始まる 1次元配列

はベクトルを表すのに用いられ,添数 1から始まる 2次元配列は行列を表すのに用いられることに注意して

下さい.「行列とベクトル」の節も参照して下さい.

2.5 レコード

Mapleは Pascalの record や Cの struct のようなレコード型データ構造を表面的には持っておりま

せん. レコード型データ構造とは,異質のオブジェクトの集まりすなわち必ずしも同一の型でないオブジェ

クトの集合を保持するデータ構造のことです.

レコード型データ構造が欲しくなる場合の 1例は,四元数を表すデータ構造を探しているときでしょう.

四元数とは, a; b; c; dを実数としたとき, a+bi+ c j+dkの形をした数です. 四元数を表示するには, 4つの

数値 a; b; c; d のみ保存すればよいのです.他の例はQ[x]内で行う多項式の因数分解を表示するデータ構造

です. a(x)の因数分解は次のようになります.

a(x) = u� f e1
1 � �� � � f en

n

ここで因子 fi 2 Q[x]の各々はモニックかつ既約です.我々は因子 fi と指数 ei と単位 u 2 Qを保持する必

要があります. Mapleにはレコードを表す方法はいくつかあります.もっとも簡単でもっともわかりやすい

方法はリストを用いることです.すなわち,四元数 a+bi+c j+dkをリスト [a; b; c; d ]で表し, a(x)の因数

分解は f を [ fi ; ei ]の形のリストのリストとして [u; f ]のリストで表すことです. このデータ構造の要素を

参照するには添数を用います. たとえば因数分解の単位元の部分は a [ 1 ] で与えられます. 次の例に示す

ように,要素を参照するのに記号を使いたければ,名前を定数と定義するマクロ (macro)機能を用いること

ができます.

> a := [ -1/2, [ [ x + 1, 2 ], [ x - 1, 1 ] ] ] ;

a :=

�
�1

2
; [ [x + 1; 2] ; [x� 1; 1] ]

�

> macro ( unit = 1, factors = 2, base = 1, exponent = 2 ) ;

> a [ unit ] ;

�1=2

> a [ factors ] [ 1 ] [ base ] ;

x + 1

> a [ unit ] � a [ factors ] [ 1 ] [ base ]ˆ a [factors ] [ 1 ] [ exponent ] �
a [ 2 ] [ 2 ] [ 1 ]ˆ a [ 2 ] [ 2 ] [ 2 ] ;

�1
2
(x + 1)2(x� 1)

Mapleでレコードを表す第 2の方法は関数呼び出しを用いることです. すなわち, a+ bi+ c j + dkを

QUARTERNION ( i, j, k, l ) で表します. この表示方法の利点は, 関数に対するいろいろな演算方法を

Mapleに伝えることができることです.これについてはあとで詳細に説明しましょう.ここでは関数をきれ

いに出力する (pretty print)方法についてだけ述べておきましょう.次の例がそれを示しております.

> QUARTERNION ( 2, 3, 0, 1 ) ;

QUARTERNION(2; 3; 0; 1)

> ‘print/QUARTERNION‘ := proc ( a, b, c, d ) a + b � ’i’ + c * ’j’ + d � ’k’ end :

> QUARTERNION ( 2, 3, 0, 1 ) ;

2 + 3i + k

ここでは出力用手続きすなわち出力用サブルーチンを定義しました. このルーチンは, Mapleから得られた

異なる QUARTERNIONごとに 1度ずつ呼び出され,その結果を表示するものです.手続きの中にクォートを

使っていることに注意して下さい.これは出力の中に名前 i; j; k; が欲しいのであって,他で何かに使われて

いる変数 i; j; k の値が欲しいわけではないからです.

レコードを表す第 3の方法は,それを体の名前で表した多変数の多項式と考え,その係数の値を保存する

ことです.これは体が数体で,その上で算術演算をしたいときにとくに役立ちます.たとえば,上で述べた出

力表示のように四元数を変数 i; j; kの多項式として表します.すなわち,

> z1 := 2 + 3 � i + k ;

z1 := 2+ 3i + k

> z2 := 2 - 3 � i + 2 � j + 2 � k ;

z2 := 2� 3i + 2 j + 2k

> coeff ( z1, i ) ; # the coefficient in i

3

> z1 + z2 ;

4 +3k + 2 j

これはすばらしいものに見えるけれども, i; j; kは繰返し文の変数として利用するので, i; j; kの名前を使う

ことは勧められません!

2.6 結合型リスト

Mapleリストは結合型リストではありません. Mapleリストは実はそのエントリのポインタの配列です.

Mapleリストと配列の違う点はリストは読み込みだけです. すなわち Mapleリストの要素には割当てがで

きません. 結合型リストは再帰的データ構造です. この差異は簡単な例題を調べればわかります. 多項式

a(x) = ∑n
i=0 ai xi を Mapleで表示することを考えましょう. 1つの方法は係数 a0; � � � ; anを Mapleリストに

保存することです.たとえば,多項式 p = x4 + 3x2 + 2x + 11を次のように表します.

[ 11, 2, 3, 0, 1 ]

多項式の次数はリストの中のエントリの (個数 �1)となります.もう 1つの別の方法は結合型リストを使用

することです.

[ 1, [ 0, [ 3, [ 2, [ 11, NIL ] ] ] ] ]

この結合型リストは再帰的データ構造であることがわかります. これは Lispプログラム言語の用語で伝統

的に CARと CDRと呼ばれている 2つの値をもつリストか,あるいは空の結合型リストを表す特別な値 NIL

のいずれかとなります. 第 1フィールドはデータの値を保持し,我々の例では係数です. 第 2フィールドは

他の結合型リストのポインタで残りのデータの値を保持します.結合型リストで表された多項式の次数を求

めるには,結合型リストの深さを計算する必要があります. pを結合型リストとすると,次の繰返し文を使っ

てこれを行うことができます.

for n from 0 while p <> NIL do p := p [ 2 ] od ;

さて, Mapleリストの代わりに結合型リストを用いて多項式を表示する理由は何でしょうか? その主た

る理由は結合型リストの先頭に新しいエントリを追加するのに,線形時間でなく一定時間で行いたいからで

す. たとえば,多項式 pに項 5x5を追加したいとしましょう. Mapleリスト表現では次のようにして 6個の

エントリをもつ新リストを作らなければなりません.

[ op( p ) , 5 ] ;

この新しいMapleリストには少なくとも 6ワードの記憶場所が必要です.こんな馬鹿なことはしないで下さ

い. op 呼び出しは op ( 1..nops ( p ) , p ) を短くしたもので, Mapleリスト p のすべてのエントリの式

列を作ります.これは一定時間で実行でき記憶場所を必要としませんが,今度は式列 ( 11, 2, 3, 0, 1 ) , 5

となり,これが作られるべきより長い新しい式列 11, 2, 3, 0, 1, 5 となります. この場合 6ワードの記憶

場所が割当てに必要となります.一般に,次数 nの多項式に an+1xn+1を追加するのに O(n)の処理時間と記

憶場所が必要です.しかし結合型リストの場合はどうでしょうか? 項 5x5を追加するには,次のようにしま

す.

[ 5 , p ] ;

これは長さが 2の新しい Mapleリストが必要となるだけなので,一定の記憶場所しか必要としません. pが

大域変数ならば線形の時間が必要となりますが, pが手続き内の局所変数やパラメータならば一定の時間し

かかかりません. この評価問題については手続きの節で説明します. いまのところ,この操作の実行時間も

結合型リストの場合 O(1)であると仮定しておきます.

Lisp プログラマーに注意 : Mapleリストの要素は変更できません. REPLACAや REPLACDR

と同じものはありません. Mapleのリスト,集合,式列,関数はデータ構造を読み込むだけです.

配列と表のみが変更可能です.

3 Mapleの手続き

3.1 引数,局所変数, RETURN, ERROR

Mapleの手続きは次の構文をとります.

proc ( nameseq)

[ local nameseq; ]

[ option nameseq; ]

statseq

end

ここで nameseqはコンマで区切られた記号の列で, statseqはセミコロンで区切られた文の列です.ここに与

えられた x; yに対して x2+y2を求める簡単な手続きがあります.

proc ( x, y ) xˆ2 + yˆ2 end

この手続きは 2つの引数 xと yを持っています.局所変数やオプションがなく, 1個の文があるだけです.こ

の手続きによって返される値は x2+y2です. 一般に手続きによって返される値は,明示的な return 文が

なければ最後に計算された値となります.明示的な return 文をもつ手続きの 1例は次のMEMBER手続き

です. MEMBER ( x, L ) は xがリスト Lの中にあれば trueを返し,なければ falseを返します.

MEMBER := proc ( x, L ) local v ;

for v in L do if v = x then RETURN (true) fi od ;

false

end ;

この MEMBER手続きは局所変数 vを持ち,大域的なユーザ変数 vと無関係です.手続きの中にでてくる変

数は引数でもなく局所変数でもなければ大域変数です.

ERROR関数は手続き内のエラーメッセージを作るのに用いられます. たとえば,次の MEMBERルーチン

は引数が実際にリストであるかどうかを検査し,そうでなければ適切なエラーメッセージを出します.

MEMBER := proc ( x, L ) local v ;

if not type ( L, list ) then ERROR ( ‘2nd argument must be a list‘ ) fi ;

for v in L do if v = x then RETURN ( true ) fi od ;

false

end ;

Maple V Release 2*2) を利用すれば,簡単に引数検査を行うもっと効率的なプログラムが書けます.

MEMBER := proc ( x, L :: list ) local v ;

for v in L do if v = x then RETURN ( true ) fi od ;

false

end ;

これはPascalやCのような型宣言ではないことに注意して下さい. これはすぐ上の MEMBERのプログラ

ムで明示的に書いた型検査を実行時に行うようにした短縮版なのです.しかもプログラムのエラーを追跡す

るときに非常に役に立つエラーメッセージも出してくれます.たとえば次のとおりです.

> MEMBER ( [ 1, x, xˆ2 ], x ) ;
Error, MEMBER expects its 2nd argument, L, to be of type list, but received x

3.2 手続きの実行のトレース : printlevel

ここにユークリッドアルゴリズムを用いて, 2つの非負の整数の最大公約数を求めるMaple手続きがあり

ます. たとえば 3は 2つの整数 21と 15を割り切る最大の整数ですから, 21と 15の最大公約数は 3です.

ところで,ユークリッド法は数学におけるもっとも古くから知られているアルゴリズムの 1つです.これは

紀元前 300年頃までさかのぼります.

GCD :=proc ( a, b ) local c,d,r ;

c := a ;

d := b ;

while d<>0 do r:=irem ( c, d ) ; c:= d ; d:= r od ;

c

end ;

irem 関数は 2つの整数の整数剰余 ( integerremainder)を計算するものです.このGCDルーチンは実際に

うまく動くでしょうか? 手続きの実行を調べるもっとも簡単な道具は printlevel 機能です. printlevel

変数は初期値として 1が割当てられている大域変数です.これをもっと高い値に設定すれば,すべての割当

て文や手続きへの入口と出口のトレースが出力されます. GCD ( 21, 15) を計算したときにどんなことが

*2)(訳注)手続き宣言の中の「 : 」が Maple V Release 4から「 :: 」に改められました.

起こるか見てみましょう.

> printlevel := 100 :

> GCD ( 21, 15 ) ;

f- -> enter GCD, args = 21, 15

c := 21

d := 15

r := 6

c := 15

d := 6

r := 3

c := 6

d := 3

r := 0

c := 3

d := 0

3

<- - exit GCD (now at top level) = 3g

3

GCD手続きへの入力引数が返される値と一緒に表示されていることがわかります. 各割当て文の実行結

果も表示されております.

MapleGCD手続きに関する興味深い点は, Mapleが任意精度の整数に関する四則演算を使っているから,

このルーチンは任意の大きさの整数に対して働くことです.たとえば, 100!と 2100の最大公約数が求められ

ます.

> GCD ( 100!, 2ˆ100 ) ;

158456325028528675187087900672

この GCD手続きは次のように再帰的にも書けるし,入力引数の型検査も含めることができます.

GCD := proc ( a :: integer, b :: integer )

if b=0 then a else GCD ( b, irem ( a, b ) ) fi

end ;

この再帰版は簡単なので理解しやすいでしょう. printlevel機能が計算の流れをどのように示すのかを見

るために,この再帰版のトレースを見てみましょう.

> GCD ( 15, 21 ) ;

f- -> enter GCD, args = 15, 21

f- -> enter GCD, args = 21, 15

f- -> enter GCD, args = 15, 6

f- -> enter GCD, args = 6, 3

f- -> enter GCD, args = 3, 0

3

<- - exit GCD (now in GCD) = 3g

3

<- - exit GCD (now in GCD) = 3g

3

<- - exit GCD (now in GCD) = 3g

3

<- - exit GCD (now in GCD) = 3g

3

<- - exit GCD (now at top level) = 3g

3

3.3 矢印演算子

数式を計算するだけの手続きの場合は,矢印構文と呼ばれる別の構文があります.これは代数でよく用い

られている関数の構文を真似したものです. 1引数の関数の場合,この構文は次のとおりです.

( symbol)�> [ local nameseq; ] expr.

引数が 0個以上あるときは,引数を括弧で囲んでおきます.

( nameseq) �> [ local nameseq; ] expr

x2+y2を計算する例は次のように書けば十分です.

( x, y ) -> xˆ2 + yˆ2 ;

Maple V Release 2*3) では,この構文は拡張されて手続き本体が if 文でもよいことになっています. した

がって区分的多項式をたとえば次のように定義することができます.

x -> if x<0 then 0 elif x<1 then x elif x<2 then 2-x else 0 fi ;

3.4 有効範囲の規則 : 引数,局所変数,大域変数

Mapleでは,入れ子になった手続きが許されています.たとえば,次のように書くことができます.

f1 := proc ( x ) local g ; g:= x �>x+1 ; x * g(x) end ;

*3)(訳注) piecewise 文を使うと x �> piecewise ( x<0, 0, x<1, x, x<2, x-2, 0 )と書けます.

手続き f1は手続きとなる局所変数 gを持っています. f1は x � (x+1)を計算しております.しかし,入れ

子型引数と局所変数は入れ子型有効範囲の規則を使用しておりません.たとえば,上の手続きは次の手続き

と同じではありません.

f2 := proc ( x ) local g ; g:=( ) �> x+1 ; x * g( ) end ;

なぜなら, gの手続き内の xの参照は, f2の引数 xを参照することにならないからです.これは大域変数 xを

参照しているのです.次の例を考えてみましょう.

> f1 ( a ) ;

a(a+1)

> f2 ( a ) ;

a(x+1)

> x := 7 ;

x := 7

> f2 ( a ) ;

8a

同様に外部の有効範囲にある局所変数を参照できません.

コンピュータ科学者への注意 : Mapleでは入れ子型手続きが許され,これは関数の値として

返されますが,入れ子型辞書式の有効範囲を認めていないので,その最終結果を直接に返すこ

とはできません. たとえば, f ->x ->f ( x+1 ) はシフト演算子を定義しておりません. これは

Mapleの将来のバージョンとみなされております.

3.5 評価規則 : 実引数と仮引数,局所変数,大域変数

関数呼び出し f ( x1, x2, � � � , xn ) を考えてみましょう.この関数呼び出しの実行は次のように行われます.

関数名 f が評価されます.次に引数 x1; x2; � � � ; xnが左から右へ評価されます.それから f が手続きであると

決まれば,この手続きが評価された引数に基づいて実行されます.この評価規則には eval, assigned, seq

を含めて 6個だけ例外があります. それでは手続き内の変数の評価はどうなるでしょうか ? 次の手続きを

考えてみましょう.

f := proc ( ) local p ; p:=xˆ2 + 4 * x + 4 ; x:=5 ; p end ;

ここで pは局所変数で,引数はありません. しかし変数 xは何でしょうか ? 「はじめに」の節で次の問題

を考えました.

> p := xˆ2 + 4 * x + 4 ;

p := x2+4x+4

> x := 3 ;

x := 3

> p ;

25

この場合は pと xは大域変数です.大域変数は完全評価すなわち再帰的に評価されるので,その結果は 25

となります.上の手続きのように pが局所変数のときはどうなるでしょうか ? すなわち,

f ( ) ;

を実行した結果どうなるでしょうか?

次の手続きのように pが引数のときはどうなるでしょうか ?

> x := ’x’ ; # Make x a symbol first

x := x

> g := proc ( p ) x:=5 ; p end :

> g ( xˆ2 + 4 * x + 4 ) ;

x2+4x+4

効率と好みの理由から, Mapleの設計者は局所変数と引数は 1レベル評価を行うことに決めました. 大域

変数のみ完全評価を行います. eval 関数は局所変数と引数を完全評価するために用いられ,大域変数の 1

レベル評価は必要なときに行うことにしております.たとえば,次のとおりです.

> x := ’x’ ; # Make x a symbol first

x := x

> g := proc ( p ) global x ; x:=5 ; eval ( p ) end :

> g ( xˆ2 +4 * x + 4 ) ;

49

3.6 漸化式とオプション記憶

フィボナッチ数 Fn は線形漸化式 F0 = 0; F1 = 1; Fn = Fn�1+Fn�2 によって定義されます.これは直接,次

のようにプログラム化されます.

F := proc(n)

if n = 0 then 0 elif n = 1 then 1 else F(n-1)+F(n-2) fi

end ;

これで最初の数個のフィボナッチ数が得られます.

> seq( F(i), i=0..10 ) ;

0;1;1;2;3;5;8;13;21;34;55

しかし,これはフィボナッチ数を求める効率的方法ではありません.実際,もっとも速いコンピュータでもこ

の手続きを用いて F(100) を求めることはできないでしょう. もし手続き Fの呼び出される回数を数える

と,同じ引数が繰返し呼び出されているのがわかるでしょう. 次の値を計算するとき,それより前の 2つの

値を記憶しておいたほうがよいのは明らかでしょう.これは次のような繰返し文で行えます.

F := proc(n) local fnm1,fnm2,f ;

if n = 0 then RETURN(0) fi ;

fnm2 := 0 ;

fnm1 := 1 ;

for i to n-1 do f := fnm1 + fnm2 ; fnm2 := fnm1 ; fnm1 := f od ;

fnm1

end ;

これをプログラム化する別の方法はオプション記憶 ( option remember)を用いることです.このオプション

は,必要な時に利用できるように,計算されたときの値を保存しておくために用いられます.次の手続きを考

えてみましょう.

F := proc(n) option remember ;

if n = 0 then 0 elif n = 1 then 1 else F(n-1)+F(n) fi

end ;

このプログラムは F(100) を素早く計算します.各Maple手続きはそれぞれ付随する記憶 (remember)表を

持っております.この表の指標は引数で,表の項目は関数の値です. nのとき Fが呼び出されると, Mapleは

まず F(n) がすでに計算されているかどうかを調べるために Fの記憶表 ( remember table)を調べます. も

しその値があれば, Fの記憶表からその結果を取り出して返します. もしその値がなければ,手続き Fのプ

ログラムを実行し,自動的に nと F(n) を対にして Fの記憶表に保存します.

いわゆる関数割当て ( functional assignment)を用いて,関数値を明示的に記憶表に保存する方法も説明し

ましておきましょう.この方法は記憶表の中に指定した値のみ保存するので,オプション記憶よりも柔軟で

す.

F := proc(n) F(n) := F(n-1)+F(n-2) end ;

F(0) := 0 ;

F(1) := 1 ;

3.7 型と一律操作

type 関数は, 入力の型に応じて違う動作をするルーチンを書くときに用います. たとえば,次の DIFF

ルーチンは与えられた変数 xの多項式で表された数式を微分するものです.

DIFF := proc( a :: algebraic, x :: name) local u,v ;

if type(a,numeric) then 0

elif type(a,name) then if a = x then 1 else 0 fi

elif type(a,‘+‘) then map( DIFF, a, x )

elif type(a,‘*‘) then u := op(1,a) ; v := a/u ;

DIFF(u,x)*v + DIFF(v,x)*u

elif type(a,anythingˆinteger) then

u := op(1,a) ; v := op(2,a) ; v*DIFF(u,x)*uˆ(v-1)

else ERROR(‘don’t know how to differentiate‘,a)

fi

end ;

DIFF の手続きの中で型は異なる 2つの目的で用いられております. 第 1の用法は型の検査です. 入力は必

ず代数的な記号式や数式でなければならないし, 微分する変数は記号名でなければなりません. 第 2の用

法は入力の型を調べて,それが和か積かに従ってどんな微分規則が適用できるかを決定することです.この

DIFF の例は非常に有用な関数である map関数の使い方を示しているので,ここでこれを説明しましょう.

これは次の構文をとります.

map( f ; a; x1; � � � ; xn )

この意味は,関数 f に補助的な引数 x1; � � � ; xnを渡して,数式 aのオペランドに関数 f を適用するものです.

上の DIFF の手続きでは 1個の補助引数 x があります. しかし,たいていは補助引数がありません. 形式的

には,これは次の列を計算するのと同値で,

seq( f( op(1,a), x1, � � � , xn ), i=1..nops(a) ) ;

この列から f の型と同じ型の値を作るものです.次に例をいくつか挙げておきます.

> p := xˆ3+2*x+1 ;

p := x3+2x+1

> map( F, p ) ;

F(x3)+F(2x)+F(1)

> map( x -> xˆ2, p ) ;

x6+4x2+1

> map( degree, p, x ) ;

4

DIFF 関数はまた構造的 (structured)な型の使い方も示しております. 型 anything, name, ‘+‘, ‘ �‘
は単純型です. 型 anything ˆ integer は構造的な型です. これは値がベキで,その底は何でもよくすな

わち任意の型でよいが,指数部は整数でなければいけないことを意味しています.これは次のように書いた

のと同値です.

if type(a,‘ˆ‘) and type(op(2,a),integer) then

構造的な型によって,長い型の検査を簡潔な検査でおきかえることができます.他のよくある例で説明しま

しょう. 多くのルーチンでは引数として,記号名や数式の集合やリストを用います. たとえば, solve コマ

ンドはいくつかの未知数に関する連立方程式を解くもので,たとえば次のようになります.

> solve ( f x+y=2, x-y=3 g, f x,y g ) ;
�

y = �1
2
;x=

5
2

�

ゼロ個以上の方程式が型 set (equation) で検査でき,ゼロ個の未知数の集合は型 set (name) で検査

できます.しかし, solve コマンドでは暗黙的にゼロに等しい代数式を解くことができます.すなわち上の

例は次のように入力することができます.

> solve ( f x+y-2, x-y-3 g, f x,y g ) ;
�

y = �1
2
; x=

5
2

�

したがって,第 1引数の型は方程式の集合か代数式の集合のいずれかで,

型 f set (algebraic), set (equation) gとなります.これは set ( f algebraic,equation g ) と

同じでないことに注意してください.なぜでしょうか?

これらと異なる型に関する情報は,オンラインヘルプから ?type で得られます.

3.8 引数の個数の変数 : args とnargs

Mapleでは関数の引数の個数を変数とする関数があります.そのような関数の 1例がmax関数です.ここ

にその関数をプログラム化した最初の試みがあります.

MAX := proc(x1) local maximum,i ;

maximum := x1 ;

for i from 2 to nargs do

if args[i] > maximum then maximum := args[i] fi

od ;

maximum

end ;

特別な変数である nargs は引数の個数で,変数 args は引数の列です. したがって args[ i ] は i 番目の

引数です.

3.9 結果を未評価で返すこと

いま書いたMAXの手続きは,数字の引数に対してのみに働きます. Maple関数max試してみれば,記号引

数でも動作することがわかるでしょう.次の文を見て下さい.

> MAX(1,2,x) ;
Error, (in MAX) cannot evaluate boolean

> max(1,2,x) ;

max(2; x)

Mapleは非数値に対して args[i] < maximum かどうかは計算できないので,手続き MAXは実行できま

せん. 数字がいくつかあるときの MAX はその答えを計算し,そうでないときは MAX(x,y) を MAX(x,y)

のままにしておきたい,そうすれば sin(x) を sin(x) のままにしておくように MAX(x,y) を記号的に

計算できます.

さらにいま書いた MAX手続きは, numeric 型の数字に対してのみ働きます. しかし, MAXが π >
p

2を

理解すれば素敵でしょう. このような MAX関数を作るのに役立つのは, 2つの実数の強力な比較を与える

signum 関数を用いることです. signum 関数は x< 0ならば �1を返し, x� 0ならば +1を返し,それ以

外は未評価 ( unevaluated)で,すなわち, signum(x)を返します.たとえば次のようになります.

> signum ( sqrt ( 2 ) -1 ) ;

1

> signum ( sqrt ( 2 ) - Pi ) ;

�1

> signum ( a - b ) ;

signum(a�b)

この signum 関数を使って, MAX関数をもっとスマートにし,記号引数も扱えるようにしましょう.

MAX := proc() local a,i,j,n,s ;

n := nargs ;

# First, put the arguments in an array

a := array(1..n) ;

for i to n do a[i] := args[i] od ;

# Compare a[i] with a[j] for 1 <= i < j <= n

i := 1 ;

while i < n do

j := i+1 ;

while j <= n do

s := signum(a[i]-a[j]) ;

if s = 1 then # i.e. a[i] >= a[j]

a[j] := a[n] ; n := n-1 ;

elif s = -1 then # i.e. a[i] < a[j]

a[i] := a[j] ; a[j] := a[n] ; j := n ; n := n-1 ; i := i-1 ;

else # cannot determine the sign

j := j+1

fi

od ;

i := i+1

od ;

if n = 1 then RETURN(a[1]) fi ;

’MAX’( seq(a[i], i=1..n) ) ;

end ;

上記のコードでもっとも面白いのは最後の行です.バッククォート ’ は実行中にMAX関数の呼び出しを防止

するもので,無限ループに落ち入らないようにしています.その代わりに未評価の関数呼び出しMAX ( . . . )

が返され,最大値が計算できなかったことを示します.しかし,若干の簡単化はたとえば次のように行ないま

す.

> MAX( x, 1, sqrt(2), x+1 ) ;

MAX (x+1;
p

2)

3.10 簡単化と変換則

代数的に記述されるすなわち変換則によって行なわれる簡単化を導入したいことがよくあります. たと

えば,関数 f が与えられていて, f が可換で結合的であるとしましょう. 実際 maxはそのような関数です.

すなわち, max(a; b) = max(b; a)で max(a; max(b; c)) = max(max(a; b); c) = max(a; b; c)が成り立ち

ます. Mapleではこれらの性質をどのように実現しているでしょうか ? 我々が欲しいのは, max関数を含

む数式を書く標準的 (canonical)方法です. 引数は整列して可換性を実現できます. 結合性については入れ

子になったmaxの呼び出しを非入れ子にします.すなわち, max(max(a;b);c)とmax(a;max(b;c))の両方を

max(a;b;c)に変換します.実際はmax(max(a)) = max(a)すなわちmaxをベキ等にしても実現できます.次

の MAX関数はその通りにしております.

MAX := proc() local a ;

a := [args] ;

a := map( flatten, a, MAX ) ; # unnest nested MAX calls

’MAX’( op(sort(a)) ) ;

end ;

flatten := proc(x,f)

if type(x,function) and op(0,x) = f then op(x) else x fi

end ;

たとえば次のようになります.

> MAX ( a, MAX ( c, b ), a ) ;

MAX (a; a; b; c)

max(a;a) = aという性質も認識できるようにすべきであることもわかります. こうするには引数をリス

トに入れる代わりに,集合に入れれば重複した要素は除かれます.また集合は自動的に整列するから,整列の

呼び出しも省略できることになります.したがって次のようになります.

MAX := proc() local a ;

a := f args g ;

a := map( flatten, a, MAX ) ;

’MAX’( op(a) );

end ;

> MAX ( a, MAX ( c, b ), a ) ;

MAX (a; b; c)

読者は我々のMAX手続きがしていることに少し当惑しているでしょう. printlevel 変数に正の整数を

割り当てれば,実行されるすべての文を追跡できることはずっと前からわかっていました. しかし,この簡

単な追跡機能を使って得られる出力は,たいていの場合多すぎます.この場合, flatten 手続きからも出力

が得られます. trace 関数はそれらと違って手続きを選別して追跡するのに用いられます. これを使って

MAX手続きを追跡してみましょう.

> trace ( MAX ) ;

MAX

> MAX ( a,MAX ( b, a ), c ) ;

f- -> enter MAX, args = b, a

a := fa; bg
a := fa; bg
MAX (a; b)

<- - exit MAX (now at top level) = MAX(a,b)g

f- -> enter MAX, args = a, MAX ( a, b ), c

a :=fa; c; MAX (a; b)g
a := fa; b; cg
MAX (a; b; c)

<- - exit MAX (now at top level) = MAX ( a, c, b )g

MAX (a; b; c)

3.11 オプション引数とデフォルト値

多くのMapleルーチンはオプション引数を受け入れます.これを用いてユーザはすべての引数の値を決める

ことはしないで,デフォルト値を使うことがよくあります.例として関数 plot, factor, collect, series

が挙げられます. degree 関数を考えてみましょう. degree 関数は 1変数の単一な多項式はその次数を求

め,多変数の多項式は次数の和を計算するもので,たとえば次のようになります.

> p := xˆ3+2*x+1 ;

p := x3+2x+1

> degree ( p ) ;

3

> q := 3*xˆ2*y+2*yˆ2-x*z+7 ;

q := 3x2y+2y2�xz+7

> degree ( q ) ;

3

ときには特定の変数たとえば xの次数を求めたいことがあるでしょう. これは degree 関数にオプション

の第 2引数を指定してできます.たとえば次のようになります.

> degree ( p, x ) ;

3

> degree ( q, x ) ;

2

この degree 関数はどのようにプログラム化するのでしょうか? 入力は数式で,第 2のオプション引数が

与えられたときは変数の名前か名前の集合であると仮定しましょう.そうすると次のように書けます.

DEGREE := proc ( a :: algebraic, x :: fname, set ( name ) g ) local s,t ;

if nargs = 1 then # determine the variable(s) for the user

s := indets(a) ; # the set of all the variables of a

if not type(s,set(name)) then ERROR(‘input not a polynomial‘) fi ;

DEGREE(a,s)

elif type(a,constant) then 0

elif type(a,name) then

if type(x,name) then if a = x then 1 else 0 fi

else if member(a,x) then 1 else 0 fi

fi

elif type(a,‘+‘) then max( seq( DEGREE(t,x), t=a ) )

elif type(a,‘*‘) then

s := 0 ;

for t in a do s := s + DEGREE(t,x) od ;

s

elif type(a,algebraicˆinteger) then DEGREE(op(1,a),x) * op(2,a)

else ERROR(‘cannot compute degree‘)

fi

end ;

ここで用いられている indets 関数は,入力に現れた未知数 (indeterminates) (すなわち変数 )の集合を

返すものです.ここで用いられている各規則を研究し,それぞれがどのような順序で実行されるかを調べる

ことを読者の宿題に残しておきましょう.

3.12 引数を通して結果を返すこと

Mapleの多くの関数は 1つ以上の値を返します. もちろん, 1つ以上の値を列やリストで返すことは常に

可能です.しかし,他のプログラミング言語と同様に引数を通して値を返すこともできて,しばしばこれが便

利になります.たとえば,多項式の長割算を行う divide 関数を考えてみましょう. divide(a,b) の呼び

出しは,多項式 bが多項式 aを割り切って剰余がないとき,そのときに限って trueを返します. たとえば次

のようになります.

> divide(xˆ3-1,x-1) ;

true

しかし通常は, aを bで割って,商 qで何かをしたいでしょう. Mapleでは次のようにこれを実行します. aを

bで割ったときの商を割り当てる記号名を,第 3引数としてたとえば次のように divide 関数に与えます.

> if divide(xˆ3-1,x-1, ’q’ ) then print(q) fi ;

x2+x+1

divide 関数には qの値でなく記号名 qを渡すために,クォートを用いていることに注意して下さい.他の

例を採り上げ,オプション引数に値を割り当てるプログラムの書き方を研究しましょう.値 xがリスト Lに

あるかどうかを検査するMEMBER関数を考えてみましょう.我々の関数を修正して, MEMBER ( x, L, ’p’ )

はリスト Lの中に xがあるかどうかを返し,さらに記号名 pには L中に最初に現れた xの位置を割り当て

るとしましょう.

MEMBER := proc ( x, L :: list,p:: name ) local i ;

for i to nops(L) do

if x = L[i] then

if nargs = 3 then p := i fi ;

RETURN ( true )

fi

od ;

false

end ;

次に例をあげましょう.

> MEMBER ( 4, [ 1, 3, 5 ], ’position’ ) ;

f alse

> position ;

posit ion

> MEMBER ( 3, [ 1, 3, 5 ], ’position’ ) ;

true

> position ;

2

MEMBER手続き内の仮引数 pの割当文の効果は,実引数 position に割当られることがわかります.

4 Mapleのプログラミング

4.1 Mapleの行列・ベクトル計算

Mapleのベクトルは添数 1から始まる 1次元配列で表わされ,行列は 1から始まる行と列の添数を持つ２

次元配列で表わされます. ここで 5�5ヒルベルト行列を作る 1つの方法をお見せしましょう. ヒルベルト

行列は対称行列で,その (i; j)要素が 1=(i + j�1)であることを思い出して下さい.

> H := array(1..5,1..5) :

> for i to 5 do for j to 5 do H[i,j] := 1/(i+j-1) od od ;

> H ;

H

> eval(H) ; 2

注意 : Hの値は行列 Hの記号名そのものであることに注意して下さい. 配列の評価規則,した

がって行列とベクトルの評価規則は特別な規則です.それは技術的な理由からです.当分の間,

行列やベクトルを出力したいときはいつも eval 関数を用いて下さい.

linalg パッケージには, Mapleでベクトルや行列を計算するための多くの関数が入っております. 上の

行列はこの線形代数パッケージの中の matrix コマンドを用いて,次のように作ることもできます.

linalg[matrix](5,5,(i,j) -> 1/(i+j-1)) ;

Mapleは行列 Hの行列式や逆行列を求めることができるし,他の行列演算も行うことができます.演算のリ

ストについては ?linalg を参照して下さい.

ガウス消去法を用いて行列のいわゆる行操作を行うプログラムをここでお見せしましょう.

GaussianElimination(A,’r’) はそれによって得られる上三角行列を計算します. これはオプション

の第 2引数も持っていて,これが与えられるとそれにその行列の階数が割当てられます.

GaussianElimination := proc( A :: matrix(rational),rank :: name)

local m,n,i,j,B,r,c,t ;

m := linalg[rowdim](A) ; # the number of rows of the matrix

n := linalg[coldim](A) ; # the number of colums of the matrix

B := array(1..m,1..n) ;

for i to m do for j to n do B[i,j] := A[i,j] od od ;

r := 1 ; # r and c are row and column indices

for c to n while r <= m do

for i from r to m while B[i,c] = 0 do od ; # search for a pivot

if i <= m then

if i <> r then # interchange row i with row r

for j from c to n do

t := B[i,j] ; B[i,j] := B[r,j] ; B[r,j] := t

od

fi ;

for i from r+1 to m do

if B[i,c] <> 0 then

t := B[i,c]/B[r,c] ;

for j from c+1 to n do B[i,j] := B[i,j]-t*B[r,j] od ;

B[i,c] := 0

fi

od ;

r := r + 1 # go to next row

fi

od ; # go to next column

if nargs>1 then rank := r-1 fi ;

eval(B)

end :

型 matrix(rational) は行列 (Mapleの 2次元配列)の要素をすべて有理数であることを指定しており

ます.

4.2 Mapleの数値計算

Mapleの浮動小数点数は小数点数として入力するか,あるいは直接に Float 関数を用いて入力します.

Float(m;e) = m�10e

ここで仮数 mは任意の大きさの整数で,指数 eはマシンサイズの整数に限定されて通常は 31ビットです.

たとえば,数値 3:1は 3.1 と入力されるか Float (31,-1) と入力されます. op 関数は浮動小数点数の仮

数や指数の抽出に用いることができます.注意 :浮動小数点定数 0.0は特別に扱われて 0と簡単化されます,

つまり自動的に整数 0になります.

ここで範囲 [0;1)内で正確に 6桁の精度の小数となる一様乱数を生成する一様乱数発生プログラムをお見

せします.

> UniformInteger := rand(0..10ˆ6-1) :

> UniformFloat := proc() Float(UniformInteger(),-6) end :

> seq( UniformFloat(), i=1..6 ) ;

:669081; :693270; :073697; :143563; :718976; :830538

Maple組込み関数 rand は与えられた範囲内の乱数を返さないことに注意して下さい. その代わりに,これ

は呼び出されたときに与えられた範囲内のランダムな整数を返すMaple手続き (乱数発生プログラム )を与

えます.

小数に丸めるときは浮動小数点演算が行われます.使用される精度は大域変数 Digits で制御され,この

デフォルト値は 10です.しかし任意の値に設定もできます. evalf 関数は正確な記号定数を浮動小数点近

似するのに用いられます.たとえば次のようになります.

> Digits := 25 :

> sin(1.0) ;

:8414709848078965066525023

> sin(1) ;

sin(1)

> evalf(%) ;

:8414709848078965066525023

Mapleは初等関数や Jv(x); Γ(x); ζ(x)のような特殊関数について知っています. Mapleではこれらはそれ

ぞれ BesselJ(v,x), GAMMA(x), Zeta(x) で表わします.これはいずれもいろいろな数列の和を計算し

高精度まで求められます.

使用される浮動小数点演算モデルでは,得られた結果の相対誤差は 101�Digits以下となります.これはハー

ドウェアで浮動小数点演算を実現している方式よりも強力で,しかもこのような精度の結果を得るにはそれ

は中間計算を高精度で行う必要があります.ここにテーラー級数の和を求める例があります.小さな xに対

して次の誤差関数

erf(x) =
2p
π

Z x

0
e�t2 dt

を計算したいとしましょう.誤差関数について良く知らなければ, Mapleでこれを少し作図してみて下さい.

小さな x;�1< x< 1に対して erf(x)を計算するには, x= 0のまわりの erf(x)のテーラー級数を用いて,

p
π

2
erf(x) =

∞

∑
n=0

(�1)nx2n+1

n! (2n+1)
= x� x3

3
+

x5

10
� x7

42
+

x9

216
��� �

(エラーチェックをしないと)次のように書けます.

ErfSmall := proc(a) local n,x,x2,result,term,sumold,sumnew ;

x := evalf(a) ; # evaluate the input at Digits precision

Digits := Digits + 2 ; # add some guard digits

sumold := 0 ;

term := x ;

sumnew := x ;

x2 := xˆ2 ;

for n from 1 while sumold <> sumnew do

sumold := sumnew ;

term := - term * x2 / n ;

sumnew := sumold + term / (2*n + 1) ;

od ;

result := evalf( 2/sqrt(Pi)*sumnew ) ;

Digits := Digits-2 ;

evalf(result) # round the result to Digits precision

end ;

大きな xに対してはこのルーチンは不正確で効率が悪くなります.大きな xに対して erf(x)を計算する方法

については練習問題を参照して下さい.

第 2の例として,整数の平方根を 20桁まで計算するニュートン反復法を書くことを考えてみましょう.方

程式 f (x) = 0を解くニュートン反復法は,

xk+1 = xk� f (xk)

f 0(xk)

で表わされることを思い出して下さい.いまの例では,
p

aを計算するために x2�a= 0を解くことになりま

す.したがって反復法は

xk+1 = (xk+
a
xk

)=2

となります.次のルーチンはこの計算を行うものです.

SqrtNewton := proc( a :: integer) local xk, xkm1 ;

if a < 0 then ERROR(‘square root of a negative integer‘) fi ;

Digits := 55 ; # add some guard digits

xkm1 := 0 ;

xk := evalf(a/2) ; # initial floating point approximation

while abs(xk-xkm1) > abs(xk)*10ˆ(-50) do

xkm1 := xk ;

print(xk) ;

xk := (xk + a/xk)/2 ;

od ;

Digits := 50 ;

evalf(xk) ; # round the result to 50 Digits

end ;

各反復法の結果を示すために print 文を入れておきました.次にそれを示しましょう.

> SqrtNewton(2) ;

1:

1:500000000000000000000000000000000000000000000000000000

1:416666666666666666666666666666666666666666666666666666

1:414215686274509803921568627450980392156862745098039216

1:414213562374689910626295578890134910116559622115744045

1:414213562373095048801689623520530243614981925776197429

1:414213562373095048801688724209698078569671875377234002

1:414213567373095048801688724209698078569671875376948073

1:4142135673730950488016887242096980785696718753769

上の例でわかるように, f 0(xk)が零に近くなく xkが f の根に十分近ければ,ニュートン反復法は 2次収束す

ることが知られています.非常に高精度すなわち Digits> 1000の場合,初期段階で最高精度になっていない

から,各反復を最高精度で行いたくないかも知れません.少し桁数を修正するだけでも反復法はうまくゆく

のでしょうか ? SqrtNewton の手続きで用いられたニュートン反復法を,各段階の桁数が倍になるように

近似的に修正してみて下さい.こうするとこの反復法はどの位速くなるでしょうか?

Mapleの浮動小数点方式はソフトウェアで実現されているので,ハードウェアの浮動小数点演算よりかな

り遅いのです. さらに Mapleはハードウェア浮動小数点演算の評価に使う evalhf と呼ばれる関数を持っ

ております. この関数は浮動小数点用の Cライブラリ組込みルーチンを使用します. だから Mapleソフト

ウェア浮動小数演算よりかなり速いのですが,これはコンパイルされないのでハードウェア浮動小数演算よ

りも遅くなります.詳細については ?evalhf を見て下さい.

4.3 Mapleの多項式計算

多項式や有理関数の計算は Mapleの長所です. ここで多項式のユークリッドノルムを計算するプログラ

ムを採り上げましょう.これは多項式 a(x) = ∑n
i=0 aixi が与えられたとき

q
∑n

i=0 a2
i を計算するものです.

EuclideanNorm := proc(a)

sqrt( convert( map( x -> xˆ2, [coeffs( expand(a) )] ), ‘+‘) )

end ;

このプログラムを内側から読んでいくと,まず入力された多項式が展開されます. coeffs 関数は係数の式

列を返すのでそれをリストにします.このリストの各要素を 2乗して新しいリストを作ります.それからこ

の 2乗リストが和に変換されます.

なぜ多項式は展開されるのでしょうか ? coeff 関数と coeffs 関数では入力多項式を必ず展開 (expanded)

するのは,そうしないと係数が計算できないからです.次の例でそれが明らかでしょう.

> p := xˆ3 - (x-3)*(xˆ2+x) + 1 ;

p := x3� (x�3)(x2+x)+1

> coeffs(p) ;
Error, invalid arguments to coeffs

> expand(p) ;

2x2+3x+1

> coeffs(expand(p)) ;

2; 3; 1

> EuclideanNorm(p) ;

p
14

注意 : 多項式のある変数に関する係数を求めたければ,常に最初にその変数について多項式を

展開したほうがよいでしょう. expand 関数を用いることができますが,これはすべての変数

について多項式展開を行うものです. 別の方法として collect 関数を使うことができます.

collect(p,x) は xに関して多項式 pを展開するものです. ?collect で他の詳細な事項

や多変数の多項式の扱い方を調べて下さい.

EuclideanNorm 手続きは数値係数の平方和の平方根を求めるという意味で,多変数の多項式に対して

動作するものです.たとえば多項式 p = ux2+y2+vが与えられれば, EuclideanNorm 手続きは
p

3を返

します.しかし,この多項式を x; yに関する多項式とみると,その係数は u;vの記号係数となります.実際は

多項式の変数が何であるかを, EuclideanNorm ルーチンに知らせることができればよいと思うでしょう.

次のように変数をオプション引数とすればそれができます.

EuclideanNorm := proc( a, v :: fname,set(name),list(name) g)
if nargs = 1 then

sqrt( convert( map( x -> xˆ2, [coeffs(expand(a))] ), ‘+‘ ) )

else

sqrt( convert( map( x -> xˆ2, [coeffs(expand(a),v)] ), ‘+‘) )

fi

end ;

型 fname, set(name), list(name) gは,第 2引数 vが単一の変数や変数の集合あるいは変数のリスト

でもよいことを示しております. coeffs 関数自身はこの引数を第 2のオプション引数として受け付けるこ

とに注意して下さい.最後にこのルーチンは入力が多項式でなくてもよいことにして,それを追加すると次

のようになります.

EuclideanNorm := proc( a, v :: fname,list(name),set(name) g)
if nargs = 1 then

if not type(a,polynom) then

ERROR(‘1st argument is not a polynomial‘,a) fi ;

sqrt( convert( map( x -> xˆ2, [coeffs(expand(a))] ), ‘+‘ ) )

else

if not type(a,polynom(anything,v)) then

ERROR(‘1st argument is not a polynomial in‘,v) fi ;

sqrt( convert( map( x -> xˆ2, [coeffs(expand(a),v)] ), ‘+‘ ) )

fi

end ;

型 polynom は次の一般的な構文

polynom(R;X)

を持っていて,これは係数が型 Rの変数 X の多項式を表しています. 1例が polynom(rational,x) で,

これは有理係数を持つ x に関する 1変数多項式すなわち Q[x]の多項式を規定しております. Rと Xが指定

されていなければ,上の最初の例のように,その数式はすべての変数に関する多項式でなければいけません.

多項式の計算を行う基本的な関数は degree, coeff, expand,divide, collect です. Mapleには

これ以外にも多項式計算を行う関数が多くあり,多項式の除算,最大公約数,終結式などの機能があります.

さらに Mapleは有限体を含むいろいろな数体の上で多項式を因子分解することもできます. 機能のリスト

は ?polynom のオンラインヘルプを参照して下さい.

有限体上で多項式計算を行う機能を説明するために, GF(2) 上で次数 nの最初の原始 3項式 (primitive

trinomial )が存在するときそれを求めるプログラムを挙げて終わりにしましょう.すなわち, xがGF(2)[x]=(a)

の原始要素であるような xn+ xm+1の形の既約多項式 aを求めたいのです. iquo 関数は 2つの整数の整

数商 (integerquotient)を計算するものです.

trinomial := proc( n :: integer ) local i,t ;

for i to iquo(n+1,2) do

t := xˆn+xˆi+1 ;

if Primitive(t) mod 2 then RETURN(t) fi ;

od ;

FAIL

end ;

4.4 手続きの読込みと保存 : read とsave

対話しながら 1行か 2行の Mapleプログラムを書けるでしょう. しかし大きなプログラムはファイルに

保存したいでしょう.普通はエディタを使ってMapleプログラムを書きファイルに保存して,それを使う前

に Mapleに読込みます. プログラムは read コマンドを用いて Mapleに読込むことができます. たとえば,

Maple手続きMAXをファイルMAXに書込んであるとすれば,次のようにしてこのファイルをMapleに読込

みます.

read MAX;

Mapleセッションで計算された Maple手続きや数式を,次の型をした save コマンドを用いて Maple内で

ファイルに保存することができます.

save f1, f2, � � � , filename ;

これは変数 f1, f2, � � � の値をテキスト形式でファイル filename に保存するものです. さらに Maple

データを内部形式すなわちいわゆる “.m ” 形式で保存することもできます.この形式は非常にコンパクトで

Mapleで読込むのが速くなります.単に次のようにファイル名に “.m ” を追加するだけです.

save f1, f2, � � � , ‘filename.m‘;

これは f1, f2, � � � の値をファイル filename.m に保存するものです.これは read コマンドを用いる

と Mapleに読み戻すことができます.

read ‘filename.m‘;

4.5 Mapleプログラムのデバッグ

もっとも簡単なデバッグ用ツールは出力レベル (printlevel)機能です. printlevel は大域変数で初期に

は 1が割当てられております. これにもっと高い値を設定すれば,すべての割当て文,手続きの入口と出口

のトレースが出力されます. printlevel の値が高ければ高いほど,それだけ手続きの深い実行レベルが

トレースされます.しかし, printlevel機能から得られる結果はしばしば大量になります. trace 関数を使う

と,指定された関数の実行のみトレースすることができます.この 2つのツールの使用例はすでにこの文書

で書きました.ここで少し詳しくお話しましょう. printlevel 変数を 3以上にすると,実行時エラーが起

きたとき, Mapleはエラーが起きた時点で呼び出し列のスタックトレースを出力します.とくに,いま実行さ

れているすべての手続きの引数およびエラーが起きた手続きで実行された局所変数の値と文を出力します.

このことをもっとも簡単に説明する例を挙げておきます.

> f := proc(x) local y ; y := 1; g(x,y) ; end :

> g := proc(u,v) local s,t ; s := 0 ; t := v/s; s+t end :

> printlevel := 4 :

> f(3) ;
f called with arguments: 3
#(f,2): g(X,Y)
g called with arguments: X, Y
#(g,2): t := v/s;

Error, (in g) division by zero
locals defined as: s = 0, t = t

4.6 他の Maple機能とのインタフェース

関数 f を手続きとしてプログラム化して, その性質を Mapleに知らせる方法をこれまで述べてきまし

た. これと違ったやり方を考えるために, f を含む数式を微分することを Mapleに教えたいとしましょ

う. f を数値的に計算して作図できるようにする方法や f を含む式を簡単化する方法などを, Mapleに教

えたいと考えてもよいでしょう. どうすればよいでしょうか ? 多くの Mapleルーチンは, これらの f 関

数に関する仕事を教えることができるようなインタフェースを持っております. そのようなルーチンは

diff, evalf, expand, combine, simplify, series などです. 関数Wを微分する方法を Mapleに

教えるには, ‘diff/W‘ と呼ばれるルーチンを書きます. この diff ルーチンがW(g)を含む式 f (x)で呼

び出されると, diff ルーチンは ‘diff/W‘(g,x) を呼び出して xに関するW(g)の導関数を計算します.

W0(x) = W(x)=(1+W(x))であることを知っているとしましょう.そうするとこの連鎖則を明示的に示す次

のような手続きを書くことができます.

‘diff/W‘ := proc(g,x) diff(g,x) * W(g)/(1+W(g)) end ;

そうすると次のようになります.

> diff(W(x),x) ;

W(x)
1+W(x)

> diff(W(xˆ2),x) ;

2
xW(x2)

1+W(x2)

2番目の例として,第 1種チェビシェフ多項式 Tn(x)の数式操作を考えましょう. この多項式は Mapleで

は T(n,x) と表すことができます.さらに特定の nの値に対して T(n,x) を xの多項式として展開したい

としましょう. Mapleの expand 関数にルーチン ‘expand/T‘ を書いてその方法を教えることができます.

expand は T(n,x) に逢うと, ‘expand/T‘(n,x) を呼出します. Tn(x)は線形漸化式 Tn(0) = 1; Tn(1) =

x; Tn(x) = 2xTn�1(x)�Tn�2 (x)を満足していることを思い出して下さい.したがって次のように書けます.

‘expand/T‘:= proc(n,x) option remember ;

if n = 0 then 1

elif n = 1 then x

elif not type(n,integer) then T(n,x) # can’t do anything

else expand(2*x*T(n-1,x) - T(n-2,x))

fi

end ;

このルーチンは再帰的ですが, rememberオプションが用いられているので T(100,x) を非常に速く計算し

ます.いくつか例を挙げましょう.

> T(4,x) ;

T(4;x)

> expand(T(4,x)) ;

8x4�8x2+1

> expand(T(100,x)) ;

1�5000x2 � � � 途中省略 � � �+633825300114114700748351602688x100

また関数 f を数値的に計算する方法を Mapleに教えるために, ‘evalf/f‘(x) が f (x)を数値的に計算で

きるようにMaple手続き ‘evalf/f‘ を定義することもできます.たとえば,前に示した数値の平方根を計

算するニュートン反復法を使いたいとしましょう.この関数は次のようになるでしょう.

‘evalf/Sqrt‘ := proc(a) local x,xk,xkm1 ;

x := evalf(a) ; # evaluate the argument in floating point

if not type(a,numeric) then RETURN( Sqrt(x) ) fi ;

if x<0 then ERROR(‘square root of a negative number‘) fi ;

Digits := Digits + 3 ; # add some guard digits

xkm1 := 0 ;

xk := evalf(x/2) ; # initial floating point approximation

while abs(xk-xkm1) > abs(xk)*10ˆ(-Digits) do

xkm1 := xk ;

xk := (xk + x/xk)/2 ;

od ;

Digits := Digits - 3 ;

evalf(xk) ; # round the result to Digits precision

end ;

> x := Sqrt(3) ;

x := Sqrt(3)

> evalf(x) ;

1:732050808

> Digits := 50 ;

Digits := 50

> evalf(Sqrt(3)) ;

1:7320508075688772935274463415058723669428052538104

4.7 外部プログラムの呼び出し

Cや Fortranで書かれた外部プログラムをMapleで呼び出したいことがしばしばあります.実際,そうした

い 2つの理由があります.明らかに, Mapleでは出来ないが他のプログラムで出来るとき, Mapleのアルゴリ

ズムを作り直す代わりに他のプログラムを利用することは意味があります. 他の理由は効率性です. Maple

は数式処理は効率が良いのですが,機械精度の数値計算は一般に効率が良くありません.そこでたとえば,固

有ベクトルを数値計算で求めるのに Fortranライブラリルーチンを利用したくなるでしょう.

Mapleのデータの受け渡しはファイルによって行わなければなりません. Mapleプログラムは外部プログラ

ムで必要とされるデータを入力ファイルとして書き出しておかなければなりません. Mapleは system コ

マンドを用いて外部プログラムを実行させることができます. 外部プログラムは入力ファイルからデータ

を呼び出し,その結果を出力ファイルに書き出さなければなりません.プログラムの実行が終了したあとで,

Mapleはその結果を出力ファイルから呼び出し, Mapleに取り込みます.こういうことを行うMapleプログ

ラミングの概略は次のようになります.

interface(quiet=true) ;

writeto(input) ;

� � � # write any data into the file input

writeto(terminal) ;

interface(quiet=false) ;

system( � � �) ; # execute the external program

read output ;

� � � # continue processing in Maple

これをもう少し詳細に調べてみましょう.最初のコマンド interface(quiet=true) はMapleのすべての

診断すなわちメッセージや警告で使用されるいろいろな出力をオフにし,それらが input ファイルに書き出

されないようにします.データが input ファイルに書かれたあとで,これは interface(quiet=false)

でリセットされます.　

次のコマンド writeto(input); は input と名付けられたファイルを出力用に開きます.この時点か

らMapleのすべての出力はこのファイルに入ります.出力は通常 lprint か printf コマンドで作られま

す.コマンド writeto はこのファイルを上書きします. すでにファイルの中にあるものにデータを追加し

たければ, writeto の代わりに appendto コマンドを使って下さい.

すべてのデータがこのファイルに書かれたあとで, コマンド writeto(terminal); を用いて出力を

ターミナルに戻すと暗黙のうちに閉じられます.

system コマンドは外部プログラムを実行するために用います. system コマンドが行う正確な呼び出

しはシステムに依存しています. Unixでは,この呼び出しは次のようになるでしょう.

system(‘foo < in > out‘) ;

外部プログラム foo は入力ファイル in 上で実行され,その結果はファイル out に書込まれます. system

コマンドは外部プログラムが正常に終了したかどうかを示す値を返します. Unixの場合,正常な終了なら 0

を返します.

最後に,データがMapleの書式で書かれていれば, read コマンドを用いて out ファイルの結果をMaple

に読込みます. 任意のデータを読込むのは以下に述べられている readline か sscanf か readdata コ

マンドを用います.

このモードでは Mapleは外部プログラムをサブルーチンのように呼び出します. 結果を Mapleに読込ん

だあとでMapleは実行を続行します.実はこれが外部プログラムを呼び出す非常に簡単な方法で,デバッグ

も簡単です. ファイルを検査して,データが決められた書式になっていることを確かめることができます.

しかし,テキストファイルを使ったデータの受け渡しはそれほど効率よい方法でないようです.多くのアプ

リケーションでは外部プログラムでなされた仕事が重要なものならば,この非効率性は問題にならないで

しょう.

4.8 数値データのファイル入出力

ここで Mapleから得たデータを他のプログラムで読めるようにファイル出力する方法や Mapleが読み込

めるように外部プログラムでデータを書式化する方法を,もっと詳細に議論しましょう.

Maple V Release 2にファイル入出力機能として,ルーチン printf, sscanf, readline があります.

ルーチン printf は一般的な書式の出力に用います. これは Cの printfルーチンにならって作られたもの

で,代数式の出力用の% aを受け付ける以外は同じです.ここで printf を用いて浮動小数点データを出力

する典型的な 1例を挙げます.

> printf(‘A float in fixed point: %f and scientific notation: %e\n‘, x, x ) ;

A float in fixed point: 0.577216 and scientific notation: 5.772157e�01

printf 関数は第 1引数として出力するべきテキストを指定する文字列を受取り, %制御列を用いてその

値はどのように出力すべきかを第 2引数として与えます.上の例では, 2つの値が% fと% e制御列で出力

されています. エスケープ列 \ nは改行するために用います. 他の役立つ制御列は整数と文字列を出力する

% dと % sで,さらに制御列 % aは Maple代数式を出力するためのものです. 他のオプションについては

?printf を参照して下さい.

readline ルーチンは,テキストファイルから 1行のテキストファイルを文字列として読むために用い

ます. sscanf ルーチンは文字列を調べて数字を抽出するために用います. printf の逆を行なうもので

す. したがってあるデータを読むためには, readline を用いてファイルの 1行ずつを読んで,各行ごとに

sscanf を用いてデータを抽出します.ユーテリティルーチン readdata を書くと空白やタブで区切られ

た数値データのカラムを持つテキストファイルを読むことができます. カラム 1個以上指定されていると,

readdata ルーチンはデータのリストやデータのリストのリストを返します.ファイル foo が次のように

なっているとしましょう.

2.1 3.2 1.2

2.4 3.1 3.4

3.9 1.1 5.5

このデータを読んだ結果がここにあります.

> readlib(readdata): # load the raaddata function from the library

> readdata(foo,2); # read the first two columns of the data

[[2:1; 3:2]; [2:4; 3:1]; [3:9; 1:1]]

4.9 FortranとCの出力

fortran コマンドと Cコマンドは Fortranや Cのコンパイラに適した書式で Mapleの数式を出力する

ものです. これは Mapleで数式や数式のベクトルや行列を開発したとき,これらの数式を Fortranや Cのサ

ブルーチンにしたいときに役立ちます.どのようにしてMapleの数式を Fortranや Cに変換するのでしょう

か? Maple関数 fortran と Cはこのためのものです. 1例を挙げてみましょう. 区間 [2,4]の 5桁の精度

で補誤差関数 erbc(x)=1-erb(x)の近似となる次の Maple多項式を作ったとしましょう.

f := �3.902704411 x + 1.890740683 � 1.714727839 x 3 + 3.465590348 x 2

� .0003861021174 x 7 + .5101467996 x 4 � .09119265524 x 5 + .009063185478 x 6

この補誤差関数がどんなものであるかはこの例題の目的にはさほど重要ではありません.実は統計学の正

規分布に関係があるのです. 近似 f をどのように作ったかを知ることもここでは重要ではありません. 我々

の Fortranと Cライブラリにはこの関数は組み込まれていないので,与えられた範囲内でこの関数のおよ

その近似を必要とします.興味のある読者は,コマンド chebyshev(erfc(x),x=2..4,10ˆ(-5)) を用

いてチェビイシェフ級数近似を作り,以前に多項式に展開するために書いた ‘expand/T‘ ルーチンを使い

ましょう. 上の近似 f を効率良く求めるために,この多項式をホーナー形式にしたのです. それから Fortran

コードをつくりたいのです.次のようになるでしょう.

> h := convert(f,horner) ;

h := 1.890740683 + (�3.902704411 + (3.465590348 + (�1.714727839 + ( .5101467996

+ (�.09119265524 + ( .009063185478� .0003861021174 x) x )x ) x ) x ) x ) x

> fortran(h) ;
t0 = 0.1890741E1+(-0.3902704E1+(0.346559E1+(-0.1714728E1+(0.510146E0+(-0.9119266E-1+(0.90631
#85E-2-0.3861021E-3�x)�x)�x)�x�)�x)�x)�x

数式は 1行より長いので, Mapleは継続記号を付け 2行の Fortranコードを生成しました.浮動小数点数は

単精度の FortranE表現に自動的に変換され, 7桁で打ち切られました. Mapleはその結果を変数 t0 に代入

しました. ここで変数 r が割当てられたファイル temp.c に Cのプログラムを出力しましょう. この C関

数は最初に Mapleに読み込まれていなければいけないことに注意して下さい.

> readlib(C) :

> C([r=h], filename=‘temp.c‘ ) ;

ファイル ‘temp.c‘ を見ると,次のようになっていることがわかります.

r = 0.1890740683E1+(�0.3902704411E1+(0.3465590348E1+(�0.1714727839E1

+(0.5101467996+(�0.9119265524E�1+(0.9063185478E�2�0.3861021174E�3

�x)�x)�x)�x)�x)�x)�x

読者はこのコマンドの追加情報機能とオプションについては, ?fortran や ?Cでヘルプページを調べて下

さい.

5 練習問題

(1) remove1(x,L) は値 xがリスト Lの中にあれば最初に出てくる xを削除し,その中になければ FAIL

を返すものです.この remove1 の Maple手続きを書きなさい.

(2)数値のリストの分散を求める variance という Maple手続きを書きなさい. variance(x) は

1
n

n

∑
i=1

(xi �µ)2

を計算するものです. ただし, nはリスト xの要素の個数で, µはリスト xの中の数値の平均です. リス

トが空ならば,ルーチンはエラーを出力して下さい.

(3) m�n行列 Aのフロベニウス・ノルムを計算するMaple手続きを書きなさい.フロベニウス・ノルムは

次のように表されます.
q

∑m
i=1 ∑n

j=1 jAi j j2

(4)「配列」の節にあったバブルソート (bublesort)アルゴリズムよりも速く数値配列を整列するアルゴリ

ズム,たとえばシェルソート,クイックソート,マージンソート,ヒープソートなどのMaple手続きを書

きなさい.そのMaple手続きを修正して,配列の 2つの要素を比較するときに用いる論理関数 f をオプ

ションの第 2引数として受け付けるようにしなさい.

(5)フィボナッチ多項式 Fn (x) を計算する Maple手続きを書きなさい. この多項式は線形漸化式 F0(x) =

1; F1(x) = x; Fn(x) = xFn�1(x)+Fn�2(x)を満足します. 最初の 10個のフィボナッチ多項式を計算し,

因子分解しなさい.そのプログラムは F50(x)を計算できますか ?

(6) Mapleの数式が与えられたとき,その数式の構造をツリーとして返す,すなわち,

� 原始的なオブジェクト (整数と文字列 )の場合はそのオブジェクトそのものを返す.

� 原始的なオブジェクトでない場合は,そのオブジェクトの型をを第 1要素とし,残りの要素はそ

のオブジェクトのオペランドの構造とするリストを返す.

とする structure という Maple手続きを書きなさい.

例 : structure(xˆ3*sin(x)/cos(x)); は次のものを返します.

[�; [ ˆ ; x; 3]; [ f unction; sin; x]; [ ˆ; [ f unction; cos; x]; �1] ]

ルーチンではMapleの型 integer, fraction, float, ‘+‘, ‘*‘, ‘ˆ ‘, string, indexed,

function, range, equation, set, list, seriese, table を扱えるようにしなさい.次

の入力であなたの関数を検査してみて下さい.

Int( exp(-t)*sqrt(t)*ln(t), t=0..infinity ) =

int( exp(-t)*sqrt(t)*ln(t), t=0..infinity ) ;

(7) DIFF 手続きを拡張して,一般なベキ,関数 ln, exp, sin, cosを微分できるようにし,与えられた入力をどの

ように微分するかわからないとき,エラーを出す代わりに未評価値を返すようにしなさい. DIFF を拡張し

て連続微分すなわち関数DIFF を DIFF するようにしなさい.たとえば, DIFF(DIFF(f(x,y),x),y)

は DIFF(DIFF(f(x,y),y),x) と同一の結果を出力するようにして下さい.

(8)数値の集合と非負の整数 nが与えられたとき,大きさ nのすべての組合せのリストを返すルーチン comb

を書きなさい.たとえば,次のようになります.

> comb( a,b,c,d, 3 ) ;

f fa;b;cg; fa;b;dg; fa;c;dg; fb;c;dg g

このルーチンを修正して,重複した要素を含むリストでも働くようにしなさい.たとえば,次のようにな

ります.

> comb( [a,b,b,c], 2 );

[ [a; b]; [a;c]; [b;b]; [b;c] ]

(9) Maple関数 degree は与えられた変数 ( s )の多項式の総次数を求めるものです.たとえば,次の多項式

では,

> p := xˆ3*yˆ3 + xˆ4 + yˆ5 :

> degree(p,x) ; # degree in x

4

> degree(p, fx,y g) # total degree in x and y

6

となります.我々はMapleでこの degree 関数をプログラム化する方法を DEGREE関数で示しました.

しかしこの degree 関数と DEGREE関数は整数のベキに対してのみ働きます.この DEGREE関数を拡

張して,一般的な数式の次数,たとえば有理数や記号式のベキに対しても働くようにしなさい.

> f := xˆ(3/2) + x + xˆ(1/2) ;

f := x
3
2 +x+x

1
2

> DEGREE(h,x) ;

3
2

> h := ( xˆn + xˆ(n-1) + xˆ2 ) * yˆm ;

h := (xn+x(n�1)+x2 )ym

> DEGREE(h,x) ;

max(n; 2)

> DEGREE(h,f x,y g) ;

max(n; 2)+m

(10)次のように表される 1変数多項式 a(x) = ∑m
i=0 ai xi と b(x) = ∑n

i=0 bi xi の和と積を計算するMaple手続

きを書きなさい.

(a) 係数の Mapleリスト [an; an�1; � � � ; a1; a0 ]

(b) 係数の結合リスト [an; [an�1; � � � ; [a1; [a0; NIL ]] � � � ]]
(11)上の問題のデータ構造では,スパースな多項式 (係数に零が多い多項式 )は零が明らさまに出てくるの

で効率が悪い.たとえば,多項式 x10+x+1を考えてみましょう.これをMapleリストとして表すと,

[10; 0; 0; 0; 0; 0; 0; 0; 1; 1]

となり,結合リストとして表すと

[10; [0; [0; [0; [0; [0; [0; [0[1[1; NILL ] ] ] ] ] ] ] ] ] ]

となります. スパース多項式のもっとも効率良い表現は非零の項のみ保存するものです. ek > ek+1 � 0

として, k番目の非零の項 ak xek を [ak; ek ]として保存することにしましょう.今度は次のように表され

たスパースな 1変量多項式の和と積を求める手続きを書いて下さい.

(a) 非零の項から成るMapleリスト

(b) 非零の項から成る結合リスト

(12)ガウス整数 Z(i)は整係数を持つ複素数,すなわち a;b2 Zで i =
p�1のとき a+biの形をした数の集

合です.このガウス整数に関するもっと興味あることは,それがユークリッド整域を形成するのかすなわ

ち素因子分解されるかということです.素因子分解できるならば,ユークリッドアルゴリズムを用いて 2

つのガウス整数のGCDを計算できるからです.ガウス整数 x = a+biが与えられたとき, jjxjj = a2+b2

を xのノルムとします. x = gy+ r; jjrjj< jjyjjとなる 2つの非零なガウス整数の剰余 r を求めるMaple

手続き REMを書いて下さい.この剰余ルーチンを用いて, 2つのガウス整数の GCDを求めなさい.

(13)Zにおける GCD計算で整数的記号的に簡単化するMapleの n変数手続き GCDを書きなさい.たとえ

ば, GCD(b,GCD(b,a),-a) が入力されたときは,このルーチンは GCD(a,b)を返します. 整数が入力

された場合はルーチンは直接 GCDを計算します.記号名が入力された場合は, GCDルーチンは次のこ

とを知っているべきです.

(a) GCD(a;b) = GCD(b;a) ( GCDは可換である )

(b) GCD(GCD(a;b;);c) = GCD(a;GCD(b;c)) = GCD(a;b;c) ( GCDは結合的である )

(c) GCD(0;a) = GCD(a) (単位元は 0である )

(d) GCD(a) = GCD(�a) = abs(x)

(14)変数リスト [v1; � � � ; vm ]に関する総次数 nの単項式のリストを求めるMaple手続きmonomial(v,n)

を書きなさい.たとえば, monomial( [ u, v ], 3) はリスト [u3; u2v; v3 ]を返します.

ヒント : t に関する積

m

∏
i=1

1
1�vi t

のテーラー級数展開を考え, tnで整理しなさい. Mapleの taylor コマンドを参照して下さい.

(15)相異なる xに対して点列 (x0; y0); (x1; y1); � � � ; (xn; yn)が与えられたとき, Mapleライブラリ関数 interp

はニュートン補間法を用いてそれらの点を補間する n次以下の多項式を求めるものです.それらの点に

対する 3次自然スプラインは区分的 3次多項式で,各区間 xi < x� xi+1は 3次多項式

fi(x) = ai x
3+bi x

2+ci x+di 1� i � n

で定義され, 4n個の未知係数は次の 4n個の条件

fi(xi�1) = yi�1; fi(xi) = yi ; i = 1; � � � ; n

f 0i (xi) = f 0i+1(xi�1); f 00i (xi) = f 00i+1(xi�1); i = 1; � � � ; n�1

f 001 (x0) = 0; f 00n (xn) = 0

によって一意に決定されます.これらの条件は得られた区分的多項式が C2連続であることを表してお

ります. 点 (xi ; yi)の入力が [x0, y0, x1, y1, � � � , xn, yn] の形のリストと変数で与えられた

とき, [f1, f2, � � � , fn] を出力するMaple手続きを書きなさい.こうするには未知係数を持つ区分

的多項式を作る必要があるので, Mapleを用いてその導関数を計算し,得られた方程式を解きます.たと

えば,次のようになります.

> spline( [ 0, 1, 2, 3 ], [0, 1, 1, 2 ], x) ;
�
�1

3
x3+

4
3

x;
2
3

x3�3x2+
13
3

x�1;�1
3

x3+3x2� 23
3

x+7

�

(16)区分的関数を表すデータ構造を設計しなさい. 前問の区分的 3次スプラインの表現は Mapleの他の機

能とインタフェースを持っておりません.そうしないで,区分的多項式を次のように未評価関数で表現

してみましょう.

IF (c1; f1; � � � ; cn�1; fn�1; fn )

は

if c1 then f1

if c2 then f2

� � �
else fn

を表すとします.

� 純粋に数値で表される条件に対する区分的多項式を簡単化する IF という Maple手続きを書き

なさい.

� IF 式が浮動小数点で評価され,その IF 式が作図できるような ‘evalf/IF‘ という手続きを書きな

さい.

� IF式を微分する ‘diff/IF‘ という手続きを書きなさい.

以上の手続きはたとえば次のような結果を出すでしょう.

> IF( x<0, sin(x), cos(x) ) ;

IF (x< 0; sin(x); cos(x))

> diff(",x) ;

IF (x< 0; cos(x); �sin(x))

> IF( Pi/3<0, sin(Pi/3), cos(Pi/3) ) ;

IF

�
1
3

Pi < 0;
1
2

3
1
2 ;

1
2

�

> evalf(") ;

:5000000000

IF 手続きを修正して,入れ子になった IF式を簡単化出来るようにして,上の例のような定数条件を扱

えるようにしなさい.たとえば,その IFは次のような結果を示すことになります.

> IF( x<0, 0, IF( x<1, xˆ2, IF(x>2, 0, 1-xˆ2) ) ) ;

IF (x< 0; 0; x< 1; x2; 2< x; 0; 1�x2 )

> IF( Pi/3<0, sin(Pi/3), cos(Pi/3) ) ;

1
2

(17)次の反復法はハートレー法として知られております.

xk+1 = xk� f (xk)= f 0(xk)=

�
1� f (xk) f 00(xk)

2 f 0(xk)2

�

Mapleでこの反復法をプログラムして,これが 3次収束することを確かめなさい.この収束は 3次であ

ることを証明しなさい.

(18)常微分方程式 (ODE)のすべての関数は閉じた形で解けなくても,コマンド dsoleve は常微分方程式

を解析的に解きます.常微分方程式

y0(x) = f (x;y(x))

と初期条件 y(0) = y0が与えられていて,級数解の最初の数項を求めたいとします.

y(x) =
∞

∑
k=0

yk xk

f (x;y(x))と初期条件 y0が入力されたとき,線形方程式を解く Maple手続きをを書きなさい.すなわち,

y(x) = y0+
n

∑
k=1

yi x
i

として,この有限和を常微分方程式に代入し,係数比較して未知数 yi について解きます. この結果を次

数 nまでに切断するには taylor コマンドを使うとよいことに注意して下さい.次の常微分方程式

y0(x) = 1+x2�2xy(x)+y(x)2; y(0) = 1

をあなたの Maple手続きで調べて,次の級数解を求めてみて下さい.

y = 1+2x+x2+x3+x4+x5+O(x6)

Mapleの dsolve コマンドを用いてこの解を解析的に求め,上の級数解と一致することをチェックして

下さい.

ニュートン反復法を用いて, 2次収束する級数 y(x)を求めなさい. k番目の近似値

xk+1 = xk� f (xk)

f 0(xk)

は O(x2k
) の精度です. したがって x0 = y0 から出発して, x2 = y0+ y1 x+ y2 x2+ y3 x3 を求めていきま

す.この級数を求める Maple手続きを書きなさい.

(19) GaussianElimination手続きを修正して,有理数から成る長方形行列について完全な行縮約を行って,こ

の行列を行縮約化階段形にして下さい.さらにその手続きを修正して要素が数式から成る行列を扱える

ようにしなさい.中間結果を簡略化するには simplify 関数を使用しなさい.

(20) erf(x)を計算するのにテーラー級数を用いると,
p

π
2

erf(x) =
∞

∑
n=0

(�1)nx2n+1

n! (2n+1)

収束が遅いから xが大きくなると次第に非効率になってきます. 正負の項が打ち消し合うために精度

も次第に悪くなってきます. 大きな xに対する erf(x)の漸近級数の項を収束するまで和をとっていく

Maple手続きを書きなさい.

p
π

2
erf(x) = 1� 2

p
π

x
e�2x

∞

∑
n=1

(�1)n1�3��� ��2n�1
2nx2n

級数を Digits の有効桁数の精度で収束させるには, xはどの位まで大きくてよいでしょうか? テー

ラー級数とこの漸近級数の両者を用いて,すべての xに対して Digits の有効桁数の精度まで erf(x)を

計算する手続きを書きなさい.


